Конвертация данных. Финальное задание

Приступаем к финалу курса.
Прием отчетов по финальному заданию продлен – до 15 декабря включительно.

Финальное задание состоит трех частей:
• Решение практической задачи
• Составление ментальной карты по важным (с Вашей точки зрения) моментам курса
• Отзыв по курсу.
Кроме этого, для успешного финала необходимо выполнить не меньше 25 домашних заданий.

Зафиналить курс – означает, что участник успешно изучил материалы курса, выполнил домашние задания (и отчитался по ним).
Финалисты первого потока курса получают от нас:
• сертификат о прохождении курса
• бесплатное участие во флешмобе по УТ 11.
У нас достаточно непросто выйти в финалисты (просто смотреть видео – этого сильно недостаточно). В среднем, с потока финалистами становятся 10-15% участников.

Скачивайте финальное задание.

Отчет о выполнении практической части оставляйте в виде комментария к этой записи.


Автор включил для этой записи групповое скрытие комментариев.

Вы сможете видеть комментарии к этой записи только после того, как оставите свой комментарий к ней и он будет одобрен администратором.

В настройках базы «Бухгалтерия предприятия» включаем опцию «Использовать дополнительные реквизиты и сведения» для доступа к механизму дополнительных реквизитов. Добавляем в «Наборы дополнительных реквизитов и сведений » реквизиты справочника «Номенклатура»: «Вес» и «Объем» числового типа. Выгружаем типовые правила конвертации объектов из баз «Бухгалтерия предприятия» и «Управление торговлей» в файлы для последующей корректировки.
Загружаем правила «УТ–БП» в конфигурацию «Конвертация данных».
В ПКС справочника «Номенклатура» добавляем параметры «Вес» и «Объем», в обработке «перед выгрузкой» – Значение = Источник.;
В ПКО, в обработке «после загрузки» добавляем

Если ПараметрыОбъекта  Неопределено Тогда
    Если ПараметрыОбъекта.Получить("Вес")  Неопределено Тогда
        Если ЗначениеЗаполнено(ПараметрыОбъекта.Вес) Тогда
            СтрокаДополнительныеРеквизиты = Объект.ДополнительныеРеквизиты.Добавить();
            Строка ДополнительныеРеквизиты.Свойство = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Вес");
            СтрокаДополнительныеРеквизиты.Значение = ПараметрыОбъекта.Вес;
        КонецЕсли;
    КонецЕсли;
    
    Если ПараметрыОбъекта.Получить("Объем")  Неопределено Тогда
        Если ЗначениеЗаполнено(ПараметрыОбъекта.Объем) Тогда
            СтрокаДополнительныеРеквизиты = Объект.ДополнительныеРеквизиты.Добавить();
            СтрокаДополнительныеРеквизиты.Свойство = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Объем");
            СтрокаДополнительныеРеквизиты.Значение = ПараметрыОбъекта.Объем;
        КонецЕсли;
   КонецЕсли;
    ОбъектМодифицирован = Истина;  
КонецЕсли;
Загружаем правила «БП--УТ» в конфигурацию «Конвертация данных».  В ПКО «Номенклатура» добавляем ПКС «Вес», «Объем»,  в обработке «перед выгрузкой»  добавляем:

Если Источник.ЭтоГруппа Тогда
    Отказ = Истина;
КонецЕсли;
 
    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    |   Номенклатура.Ссылка,
    |   Номенклатура.Свойство,
    |   Номенклатура.Значение
    |ИЗ
    |   Справочник.Номенклатура.ДополнительныеРеквизиты КАК Номенклатура
    |       ВНУТРЕННЕЕ СОЕДИНЕНИЕ (ВЫБРАТЬ
    |           ДополнительныеРеквизитыИСведения.Ссылка КАК Ссылка,
    |           ДополнительныеРеквизитыИСведения.ВерсияДанных КАК ВерсияДанных,
    |           ДополнительныеРеквизитыИСведения.ЭтоДополнительноеСведение КАК ЭтоДополнительноеСведение
    |       ИЗ
    |           ПланВидовХарактеристик.ДополнительныеРеквизитыИСведения КАК ДополнительныеРеквизитыИСведения
    |       ГДЕ
    |           ДополнительныеРеквизитыИСведения.Наименование ПОДОБНО ""Вес"") КАК Реквизит
    |       ПО Номенклатура.Свойство = Реквизит.Ссылка
    |ГДЕ
    |   Номенклатура.Ссылка = &amp;Ссылка";
    Запрос.УстановитьПараметр("Ссылка", Источник.Ссылка);

    Результат = Запрос.Выполнить().Выбрать();
    Если Результат.Следующий() Тогда
        Значение = Результат.Значение;
    Иначе
        Значение = 0;
    КонецЕсли;
Выгружаем исправленные правила в файлы, загружаем в планы обмена соответствующих конфигураций, заменяем типовые правила на доработанные.

Добрый день.
Устанавливаем типовые Бух 3 и УТ 11.
В бухгалтерии влючаем дополнительные реквизиты в настройках программы.
Выгружаем из УТ11 типовые правила обмена.
Загружаем в конвертацию.
Создаем ПКО «ДополнительныеРеквизитыИСведения_без_источника» без источника, одно ПКС «Наименование», событие «Перед выгрузкой»

Значение=Источник;
Создаем ПКО «ЗначенияСвойствОбъектов_без_источника» без источника,
ПКС «Наименование» событие «Перед выгрузкой»

Значение=Источник;
и «Владелец» галочка получить из входящих данных.
Находим ПКО «Номенклатура» добавляем ПКГС «ДополнительныеРеквизиты» источник пустой. Событие «Перед обработкой»

КоллекцияОбъектов = Новый ТаблицаЗначений;
    КоллекцияОбъектов.Колонки.Добавить("Значение");
    КоллекцияОбъектов.Колонки.Добавить("Свойство");
    КоллекцияОбъектов.Колонки.Добавить("ТекстоваяСтрока");
    
    СтрокаКоллекцияОбъектов = КоллекцияОбъектов.Добавить();
    СтрокаКоллекцияОбъектов.Свойство = "Вес";
    СтрокаКоллекцияОбъектов.Значение = Источник.Вес;      
    СтрокаКоллекцияОбъектов.ТекстоваяСтрока = "";
    СтрокаКоллекцияОбъектов = КоллекцияОбъектов.Добавить();
    СтрокаКоллекцияОбъектов.Свойство = "Объем";
    СтрокаКоллекцияОбъектов.Значение = Источник.Объем;      
    СтрокаКоллекцияОбъектов.ТекстоваяСтрока = "";
ПКС «Свойство» правило конвертации «ДополнительныеРеквизитыИСведения_без_источника», ПКС «Значение» равило конвертации «ЗначенияСвойствОбъектов_без_источника», событие перед выгрузкой

Если ОбъектКоллекции.Свойство = "Вес" Тогда
    ИсходящиеДанные = Новый Структура;
    ИсходящиеДанные.Вставить("Владелец","Вес");
ИначеЕсли ОбъектКоллекции.Свойство = "Объем" Тогда
    ИсходящиеДанные = Новый Структура;
    ИсходящиеДанные.Вставить("Владелец","Объем");
КонецЕсли;
Сохраняем правила, загружаем правила конвертации объектов нашего обмена данными в УТ11, проверяем, перенос осуществляется.

Установил УТ версии 11.0.9.14
Установил бухгалтерию версии 3.0.16.9
В бухгалтерии включил использование дополнительных реквизитов и сведений.

Настройка выгрузки реквизитов в дополнительные реквизиты очень похожа на прошлую работу, но в новой версии конфигурации изменился принцип хранения дополнительных реквизитов. Сейчас информация о типах хранится в дополнительном справочнике ЗначенияСвойствОбъектов, а сами данные хранятся в табличных частях самих справочников.
Кратко:
Не стал создавать служебных ПКО для переноса значений реквизитов.
Обновил метаданные обеих конфигураций. ПКГС «ДополнительныеРеквизиты» в ПКО «Номенклатура». Далее стан настраивать перенос «Свойство», и «Значение» через ПКО: «ДополнительныеРеквизитыИСведения» и «ЗначенияСвойствОбъектов» соответственно.
В событии выгрузки свойства наладил следующий код:

ИсходящиеДанные = Новый Структура("ТипЗначения");
ИсходящиеДанные.ТипЗначения = "СправочникСсылка.ЗначенияСвойствОбъектов";
ИсходящиеДанные.Вставить("ЭтоДопОбработка",Истина);

ИсходящиеДанные.Вставить("Комментарий","");
ИсходящиеДанные.Вставить("МногострочноеПолеВвода",0);
ИсходящиеДанные.Вставить("Подсказка","");
ИсходящиеДанные.Вставить("ПометкаУдаления",Ложь);
ИсходящиеДанные.Вставить("Родитель", ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.ПустаяСсылка());
ИсходящиеДанные.Вставить("СклоненияПредмета","");
ИсходящиеДанные.Вставить("ФорматСвойства","");
ИсходящиеДанные.Вставить("ЭтоГруппа",Ложь);
ИсходящиеДанные.Вставить("ЭтоДополнительноеСведение",ложь);

Значение = ОбъектКоллекции.Свойство;

// Сформируем данные для таблицы НаборыДополнительныхРеквизитовИСведений
ДанныеТЗ = Новый Структура("Наименование, ЭтоГруппа, ДополнительныеРеквизиты, ДополнительныеСведения");

ДанныеТЗ.Наименование = "Справочник ""Номенклатура""";
ДанныеТЗ.ЭтоГруппа = Ложь;

ДопРеквизиты = Новый ТаблицаЗначений;
ДопРеквизиты.Колонки.Добавить("Свойство");
ДопРеквизит = ДопРеквизиты.Добавить();
ДопРеквизит.Свойство = ОбъектКоллекции.Свойство; 

ДанныеТЗ.ДополнительныеРеквизиты = ДопРеквизиты;
ДанныеТЗ.ДополнительныеСведения = Новый ТаблицаЗначений;

ВыгрузитьПоПравилу(ДанныеТЗ, , ИсходящиеДанные, , "НаборыДополнительныхРеквизитовИСведений",,,,,,,,Ложь);
В ПКС «Значение», код:

ИсходящиеДанные = Новый Структура;
ИсходящиеДанные.Вставить("Владелец",ОбъектКоллекции.Свойство);

//Сообщить("Выгрузка наименования значения = "+Строка(ОбъектКоллекции.Значение)+ " Тип: "+ ТипЗнч(Строка(ОбъектКоллекции.Значение)));
ИсходящиеДанные.Вставить("Наименование", Строка(ОбъектКоллекции.Значение));
ИсходящиеДанные.Вставить("Вес",0);   


ИсходящиеДанные.Вставить("ТипЗначения","СправочникСсылка.ЗначенияСвойствОбъектов");
ИсходящиеДанные.Вставить("ЭтоДопОбработка",Истина);  
ИсходящиеДанные.Вставить("Комментарий","");
ИсходящиеДанные.Вставить("МногострочноеПолеВвода",0);
ИсходящиеДанные.Вставить("Подсказка","");
ИсходящиеДанные.Вставить("ПометкаУдаления",Ложь);
ИсходящиеДанные.Вставить("Родитель", ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.ПустаяСсылка());
ИсходящиеДанные.Вставить("СклоненияПредмета","");
ИсходящиеДанные.Вставить("ФорматСвойства","");
ИсходящиеДанные.Вставить("ЭтоГруппа",Ложь);
ИсходящиеДанные.Вставить("ЭтоДополнительноеСведение",ложь);

ИмяПКО = "ЗначенияСвойствОбъектов";
Создал ПКО «НаборыДополнительныхРеквизитовИСведений», заполнял ее передавая источник из ПКС «Свойство» при конвертации номенклатуры.

В целом все корректно переносится, за исключением, автоматически не заполняется табличная часть НаборыДополнительныхРеквизитовИСведений соответсвующими свойствами объекта. А при открытии номенклатуры описания заголовка дополнительных свойств были зачеркнуты. Остальные данные успешно переносятся.

В конфигурации Бухгалтерия 3.0 в Настройках программы устанавливаем использование дополнительных реквизитов и сведений.
Далее в администрировании открываем Набооры дополнительных реквизитов и сведений и для справочника номенклатура устанавливаем ревизиты Вес и Объем числового типа.
Теперь у нас в Номенклатуре появились нужные нам дополнительные реквизиты, на которые будем переносить данные.

В правилах УТ-БУ модифицируем спр.Номенклатура. Добавляем параметры Вес и Объем с текстом ПКС Перед выгрузкой:

Если Источник.ЭтоГруппа Тогда 
    Отказ = Истина;
Иначе
    Значение = Источник.Вес;
    Отказ = Значение = 0;   
КонецЕсли;
и в ПКО Номенклатура После загрузки модифицируем код, добавляя в конец кода текст

Если ПараметрыОбъекта <> Неопределено Тогда
    // загружаем вес
    Вес = ПараметрыОбъекта.Получить("Вес");
    Если Вес <> Неопределено И Параметры.ДопРеквизитВес <> Неопределено Тогда
        СтрокаДР = Объект.ДополнительныеРеквизиты.Найти(Параметры.ДопРеквизитВес, "Свойство");
        Если СтрокаДР = Неопределено Тогда
            СтрокаДР = Объект.ДополнительныеРеквизиты.Добавить();
            СтрокаДР.Свойство = Параметры.ДопРеквизитВес;            
        КонецЕсли;
        СтрокаДР.Значение = Вес;
        ОбъектМодифицирован = Истина;
    КонецЕсли;
    
    // загружаем объем
    Объем = ПараметрыОбъекта.Получить("Объем");
    Если Объем <> Неопределено И Параметры.ДопРеквизитОбъем <> Неопределено Тогда
        СтрокаДР = Объект.ДополнительныеРеквизиты.Найти(Параметры.ДопРеквизитОбъем, "Свойство");
        Если СтрокаДР = Неопределено Тогда
            СтрокаДР = Объект.ДополнительныеРеквизиты.Добавить();
            СтрокаДР.Свойство = Параметры.ДопРеквизитОбъем;        
        КонецЕсли;
        СтрокаДР.Значение = Объем;
        ОбъектМодифицирован = Истина;  
    КонецЕсли;
КонецЕсли;
А также в правилах конвертации в событии Перед загрузкой данных добавляем в конец кода определение свойств параметров

Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|   ДопРеквизиты.Свойство
|ИЗ
|   Справочник.НаборыДополнительныхРеквизитовИСведений.ДополнительныеРеквизиты КАК ДопРеквизиты
|ГДЕ
|   ДопРеквизиты.Ссылка = ЗНАЧЕНИЕ(Справочник.НаборыДополнительныхРеквизитовИСведений.Справочник_Номенклатура)
|   И ДопРеквизиты.Свойство.Наименование ПОДОБНО ""Вес (нетто)""";

Результат = Запрос.Выполнить().Выбрать();
Если Результат.Следующий() Тогда
    Параметры.Вставить("ДопРеквизитВес", Результат.Свойство);
Иначе
    Параметры.Вставить("ДопРеквизитВес", Неопределено);
КонецЕсли;

Запрос.Текст =
"ВЫБРАТЬ
|   ДопРеквизиты.Свойство
|ИЗ
|   Справочник.НаборыДополнительныхРеквизитовИСведений.ДополнительныеРеквизиты КАК ДопРеквизиты
|ГДЕ
|   ДопРеквизиты.Ссылка = ЗНАЧЕНИЕ(Справочник.НаборыДополнительныхРеквизитовИСведений.Справочник_Номенклатура)
|   И ДопРеквизиты.Свойство.Наименование ПОДОБНО ""Объем""";

Результат = Запрос.Выполнить().Выбрать();
Если Результат.Следующий() Тогда
    Параметры.Вставить("ДопРеквизитОбъем", Результат.Свойство);
Иначе
    Параметры.Вставить("ДопРеквизитОбъем", Неопределено);
КонецЕсли;
Выгружаем правила, проверяем при помощи загрузки правил из файла в обмены.

Финальное задание (практическая часть)
Обмен с использованием типовых правил между УТ и БП был настроен в одном из домашних заданий по курсу. Осталось только модифицировать правила.
1. В БП в настройках программы включила использование Дополнительных реквизитов.
2. Из УТ выгрузила правила конвертации и загрузила их в КД.
3. В ПКО Номенклатура добавила ПКГС ДополнительныеРеквизиты.
4. В обработчике ПередОбработкой прописала:

Если Не Источник.ЭтоГруппа Тогда
    
    Запрос = Новый Запрос;
    Запрос.Текст = "ВЫБРАТЬ
    |   ""Вес"" КАК Свойство,
    |   Номенклатура.Вес КАК Значение,
    |   """" КАК ТекстоваяСтрока
    |ИЗ
    |   Справочник.Номенклатура КАК Номенклатура
    |ГДЕ
    |   Номенклатура.Ссылка = &Ссылка
    |
    |ОБЪЕДИНИТЬ ВСЕ
    |
    |ВЫБРАТЬ
    |   ""Объем"",
    |   Номенклатура.Объем,
    |   """" КАК ТекстоваяСтрока
    |ИЗ
    |   Справочник.Номенклатура КАК Номенклатура
    |ГДЕ
    |   Номенклатура.Ссылка = &Ссылка";
    
    Запрос.УстановитьПараметр("Ссылка", Источник.Ссылка);
    
    РезультатЗапроса = Запрос.Выполнить();
    КоллекцияОбъектов = РезультатЗапроса.Выгрузить();
КонецЕсли;
5. Сохранила и загрузила правила в план обмена УТ – БП (в базе УТ). Выполнила синхронизацию.

Подготавливаем демо-базы с последними версиями бухгалтерии и управления торговлей. Открываем в конфигураторе базу торговли и сохраняем в файл правила обмена с бухгалтерией. Также открывем торговлю в режиме предприятия и сохраняем информацию о метаданных в файл.

В конвертации данных загружаем правила обмена данными, а затем обновляем информацию о конфигурации торговли, так как реквизиты «Вес» и «Объем» в типовой обмен не входят.

Вдумчиво просматривая типовые правила конвертации, мы видим, что реализован механизм переноса в дополнительные сведения данных по договорам. Вроде бы и то, но не совсем: дополнительные сведения хранятся в регистре сведений, а дополнительнные реквизиты, куда нам нужно перенести вес и объем, хранятся в табличной части справочника. Но вот часть кода, отвечающего за перенос планов видов характеристик можно использовать, не городя лишних велосипедов.

Итак, находим существующее ПКО «СвойстваОбъектов». Оно получает на вход структуру, а на выходе выдает план видов характеристик. Смотрим список ПКС: все, кроме наименования задается программно. Первым меняем ПКС «ЭтоДополнительноеСведение». Удаляем программную установку значения, так как для номенклатуры там нужно будет передавать дополнительные реквизиты. Далее, ПКС «НазначениеСвойства». Помимо свойств договоров у нас теперь будут и реквизиты номенклатуры, так что удаляем жесткое определение значения программным образом и в обработчик ПКС «Перед обработкой» добавляем следующий код:

Значение = Источник.НазначениеСвойства;
Теперь займемся обработчиками ПКО «СвойстваОбъектов». Мы выгружаем план видов характеристик, а значит нужны дополнительные правки. Обработчик «При выгрузке» приводим к такому виду:

УзелТипов = СоздатьУзел("Типы");
Если (Источник.Наименование = "Вес") ИЛИ (Источник.Наименование = "Объем") Тогда
    Если Источник.Наименование = "Вес" Тогда
        УстановитьАтрибут(УзелТипов, "Разрядность", 15);
        УстановитьАтрибут(УзелТипов, "РазрядностьДробнойЧасти", 3);
    Иначе
        УстановитьАтрибут(УзелТипов, "Разрядность", 18);
        УстановитьАтрибут(УзелТипов, "РазрядностьДробнойЧасти", 6);
    КонецЕсли;
    одЗаписатьЭлемент(УзелТипов, "Тип", "Число");
Иначе
    одЗаписатьЭлемент(УзелТипов, "Тип", "СправочникСсылка.ЗначенияСвойствОбъектов");
КонецЕсли;
ДобавитьПодчиненный(Приемник, УзелТипов);
Теперь нам нужно поправить все участки, ссылающиеся на ПКО «СвойстваОбъектов». Благо ходить недалеко и все объекты лежат в той же группе договоров. Для всех ПКС близлежащих объектов, использующих правило «СвойстваОбъектов», в обработчике «Перед обработкой» после куска кода

Значение = Новый Структура;
Значение.Вставить("Ссылка");
Значение.Вставить("Наименование", "Чего-то там из УТ11");
добавляем

Значение.Вставить("НазначениеСвойства", "Справочник_ДоговорыКонтрагентов");
Значение.Вставить("ЭтоДополнительноеСведение", Истина);
Теперь переходим к нашей непосредственной цели – справочнику номенклатуры. Открываем соответствующее ПКО и делаем примерно то же самое, что делается в типовых правилах с договорами: перенос свойств и их значений через параметры. Даже не смотря на то, что дополнительные реквизиты хранятся в табличной части этого же справочника, создавать ПКГС нет смысла, т.к. в базе-приемнике могут быть заведены другие дополнительные реквизиты и все равно придется писать код для того, чтобы не затереть лишнего.

Итак, добавляем ПКС с реквизитом-источником «Вес» и передачей в параметр «ЗначениеСвойстваВес». Аналогично, для реквизита «Объем» делаем передачу в параметр «ЗначениеСвойстваОбъем». Затем добавляем ПКС с пустым источником и передачей в параметр «СвойствоВес», значение заполняем программно в обработчике «Перед обработкой»:

Значение = Новый Структура;
Значение.Вставить("Ссылка");
Значение.Вставить("Наименование", "Вес");
Значение.Вставить("НазначениеСвойства", "Справочник_Номенклатура");
Значение.Вставить("ЭтоДополнительноеСведение", Ложь);
ВыгрузитьТолькоСсылку = Ложь;
Аналогично и для ПКС с передачей в параметр «СвойствоОбъем»:

Значение = Новый Структура;
Значение.Вставить("Ссылка");
Значение.Вставить("Наименование", "Объем");
Значение.Вставить("НазначениеСвойства", "Справочник_Номенклатура");
Значение.Вставить("ЭтоДополнительноеСведение", Ложь);
ВыгрузитьТолькоСсылку = Ложь;
Обработчики, отвечающие за выгрузку, готовы. Теперь переходим к следующей части – загрузке.

Открываем ПКО «Номенклатура» и добавляем в обработчик «После загрузки» следующий код:

// Перенос свойств номенклатуры
Если РежимЗагрузкиДанныхВИнформационнуюБазу() И НЕ Объект.ЭтоГруппа
 Тогда
    
    ТабЧасть = Объект.ДополнительныеРеквизиты;
    
    Для Каждого ПараметрОбъекта Из ПараметрыОбъекта Цикл
        
        Ключ = ПараметрОбъекта.Ключ;
        Если Лев(Ключ, 16) = "ЗначениеСвойства" Тогда
            
            Свойство = ПараметрыОбъекта[СтрЗаменить(Ключ, "ЗначениеСвойства", "Свойство")];
            
            СтрокаРеквизита = ТабЧасть.Найти(Свойство.Ссылка, "Свойство");
            Если СтрокаРеквизита = Неопределено Тогда
                СтрокаРеквизита = ТабЧасть.Добавить();
                СтрокаРеквизита.Свойство = Свойство.Ссылка;
                СтрокаРеквизита.Значение = ПараметрОбъекта.Значение;
                ОбъектМодифицирован = Истина;
            ИначеЕсли СтрокаРеквизита.Значение <> ПараметрОбъекта.Значение Тогда
                СтрокаРеквизита.Значение = ПараметрОбъекта.Значение;
                ОбъектМодифицирован = Истина;
            КонецЕсли;
            
        КонецЕсли;
    КонецЦикла;
КонецЕсли;
Тут есть два момента: во-первых, реквизит «ОбъектМодифицирован» изменяется в уже существующем коде этого обработчика, поэтому его нужно устанавливать, если изменяем элемент справочника. Во-вторых, вставка заключена в условие с использованием функции «РежимЗагрузкиДанныхВИнформационнуюБазу», что является в данном случае необходимым для online-обмена, так как при первом интерактивном обмене не будут грузиться данные по ссылкам и обращение по ссылке к загруженному элементу плана видов характеристик вызовет ошибку.

Теперь займемся механизмом постобработки планов видов характеристик. Открываем саму конвертацию и в обработчике «Перед загрузкой данных» находим вот такой фрагмент кода

ЗагруженныеСвойства = Новый ТаблицаЗначений;
ЗагруженныеСвойства.Колонки.Добавить("СвойствоСсылка", Новый ОписаниеТипов("ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения"));
Несколькими шагами ранее мы сделали ПКО «СвойстваОбъектов» более универсальным, теперь оно подходит не только для договоров, но и для номенклатуры. Добавим еще одну колонку в служебную таблицу для идентификации нужного нам справочника

ЗагруженныеСвойства.Колонки.Добавить("НазначениеСвойства", Новый ОписаниеТипов("Строка",, Новый КвалификаторыСтроки(50)));
Затем идем в обработчик «После загрузки данных» и видим, что за дальнейшую обработку планов видов характеристик отвечают два алгоритма: «ОбработкаПлановВидовХарактеристикОтложенная» и «ОбновитьНаборыДополнительныхРеквизитовИСведений».

Сначала займемся алгоритмом «ОбработкаПлановВидовХарактеристикОтложенная». Данный алгоритм выбирает из списка загруженных объектов планы видов характеристик, перенесенные по правилу «СвойстваОбъектов» и сохраняет в таблице значений для дальнейшего использования, приведем его вот к такому виду:

Если ИмяПКО = "СвойстваОбъектов" Тогда
    
    Если РежимЗагрузкиДанныхВИнформационнуюБазу() Тогда
        
        ЗагруженныеСвойстваДоговоров = Параметры.ЗагруженныеСвойства;
        
        НоваяСтрока = ЗагруженныеСвойстваДоговоров.Добавить();
        НоваяСтрока.СвойствоСсылка = Объект.Ссылка;
        НоваяСтрока.НазначениеСвойства = ПараметрыОбъекта["НазначениеСвойства"]; // <--- указываем, к какому справочнику относится свойство
        
        ОбъектМодифицирован = Ложь;
        
    КонецЕсли;
    
КонецЕсли;
Теперь переходим к алгоритму «ОбновитьНаборыДополнительныхРеквизитовИСведений». Данный алгоритм, используя таблицу значений, созданную предыдущим алгоритмом, добавляет в табличные части предопределенных элементов справочника «НаборыДополнительныхРеквизитовИСведений» перенесенные элементы плана видов характеристик «ДополнительныеРеквизитыИСведения». Изначально он рассчитан на дополнительные свойства справочника договоров, дорабатываем его для дополнительных реквизитов справочника номенклатуры:

ЗагруженныеСвойства = Параметры.ЗагруженныеСвойства;
ЗагруженныеСвойства.Свернуть("СвойствоСсылка,НазначениеСвойства");

Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|   СоставТребуемый.СвойствоСсылка КАК Ссылка,
|   СоставТребуемый.НазначениеСвойства КАК НазначениеСвойства
|ПОМЕСТИТЬ СоставТребуемый
|ИЗ
|   &СоставТребуемый КАК СоставТребуемый
|
|ИНДЕКСИРОВАТЬ ПО
|   Ссылка";
Запрос.УстановитьПараметр("СоставТребуемый", ЗагруженныеСвойства);
Запрос.МенеджерВременныхТаблиц = Новый МенеджерВременныхТаблиц;
Запрос.Выполнить();

ШаблонЗапроса =
"ВЫБРАТЬ
|   СоставТребуемый.Ссылка
|ИЗ
|   СоставТребуемый КАК СоставТребуемый
|       ЛЕВОЕ СОЕДИНЕНИЕ Справочник.НаборыДополнительныхРеквизитовИСведений._ИмяТЧ_ КАК СоставТекущий
|       ПО СоставТребуемый.Ссылка = СоставТекущий.Свойство
|           И (СоставТребуемый.Ссылка = ЗНАЧЕНИЕ(Справочник.НаборыДополнительныхРеквизитовИСведений._НазначениеСвойства_))
|ГДЕ
|   СоставТекущий.Ссылка ЕСТЬ NULL
|   И СоставТребуемый.НазначениеСвойства = ""_НазначениеСвойства_""";

Запрос.Текст = СтрЗаменить(СтрЗаменить(ШаблонЗапроса, "_ИмяТЧ_", "ДополнительныеСведения"), "_НазначениеСвойства_", "Справочник_ДоговорыКонтрагентов");
РезультатЗапроса = Запрос.Выполнить();

Если Не РезультатЗапроса.Пустой() Тогда
    
    НаборСведенийОбъект = Справочники.НаборыДополнительныхРеквизитовИСведений.Справочник_ДоговорыКонтрагентов.ПолучитьОбъект();
    ТабДополнительныеСведения = НаборСведенийОбъект.ДополнительныеСведения;
    
    ВыборкаСвойств = РезультатЗапроса.Выбрать();
    Пока ВыборкаСвойств.Следующий() Цикл
        
        НоваяСтрока = ТабДополнительныеСведения.Добавить();
        НоваяСтрока.Свойство = ВыборкаСвойств.Ссылка;
        
    КонецЦикла;
    
    НаборСведенийОбъект.ОбменДанными.Загрузка = Истина;
    НаборСведенийОбъект.Записать();
    
КонецЕсли;

Запрос.Текст = СтрЗаменить(СтрЗаменить(ШаблонЗапроса, "_ИмяТЧ_", "ДополнительныеРеквизиты"), "_НазначениеСвойства_", "Справочник_Номенклатура");
РезультатЗапроса = Запрос.Выполнить();

Если Не РезультатЗапроса.Пустой() Тогда
    
    НаборСведенийОбъект = Справочники.НаборыДополнительныхРеквизитовИСведений.Справочник_Номенклатура.ПолучитьОбъект();
    ТабДополнительныеРеквизиты = НаборСведенийОбъект.ДополнительныеРеквизиты;
    
    ВыборкаСвойств = РезультатЗапроса.Выбрать();
    Пока ВыборкаСвойств.Следующий() Цикл
        
        НоваяСтрока = ТабДополнительныеРеквизиты.Добавить();
        НоваяСтрока.Свойство = ВыборкаСвойств.Ссылка;
        
    КонецЦикла;
    
    НаборСведенийОбъект.ОбменДанными.Загрузка = Истина;
    НаборСведенийОбъект.Записать();
    
КонецЕсли;
Еще один маленький нюанс: в управлении торговлей реквизит «Объем» имеет тип Число(18,6), а в бухгалтерии максимальная разрядность, определенная для числа в плане видов характеристик, составляет (15,3). Так что открываем базу бухгалтерии в конфигураторе и правим описание типов у плана видов характеристик.

Ну вот теперь можно выгружать правила и проверять обмен данными. Если нигде не допустили ошибок, все должно заработать.

- Включил использование доп. реквизитов в конфигурации БП – «Использовать доп. реквизиты».
- Добавил «Вес» и «Объем» в доп. реквизиты «НоменклатурЫ»
- загрузил в КД правила обмена УТ11-БП30.
- Добавил новое ПКО для справочника «Номенклатура»(с пустым источником) добавил пкгс для ТЧ «ДополнительныеРеквизиты» и установил галки получения из входящих данных.
- Создал ПКО для ДопРеквизитыИСведения (пвх), с пкс наименование В обработчике «Перед выгрузкой» определил
Значение = Источник;
- Добавил ПКО ЗначенияСвойствОбъектовМоё (справочник) с пкс Наименование;
- ПКС Владелец указал значение, в зависимости от свойства.
- Добавил ПВД Вес с произвольным алгоритмом, с пко Свойства

и перед обработкой написал:
Запрос = Новый Запрос;
Запрос.Текст =
«ВЫБРАТЬ
| Номенклатура.Ссылка,
| Номенклатура.Вес,
| «»Вес»" КАК Свойство
|ИЗ
| Справочник.Номенклатура КАК Номенклатура
|ГДЕ
| Номенклатура.Вес «»"»";

Результат = Запрос.Выполнить();

ВыборкаСсылка = Результат.Выгрузить();

Пока ВыборкаСсылка.Следующий() Цикл

ИсходящиеДанные = Новый Структура;
Ссылка = ВыборкаСсылка.Ссылка;
ИсходящиеДанные.Вставить(«ДополнительныеРеквизиты», Новый ТаблицаЗначений);

ДополнительныеРеквизиты = ИсходящиеДанные.ДополнительныеРеквизиты;
ДополнительныеРеквизиты.Колонки.Добавить(«Значение»);
ДополнительныеРеквизиты.Колонки.Добавить(«Свойство»);

Реквизит = ДополнительныеРеквизиты.Добавить();
Реквизит.Значение = ВыборкаСсылка.Вес;
Реквизит.Свойство = ВыборкаСсылка.Свойство;

ВыгрузитьПоПравилу(Ссылка, , ИсходящиеДанные, , «Свойства»);

КонецЦикла;
8) Создал пвд Объем с произвольным алгоритмом, с пко Свойства и перед обработкой написал:
Запрос = Новый Запрос;
Запрос.Текст =
«ВЫБРАТЬ
| Номенклатура.Ссылка,
| Номенклатура.Объем,
| «»Объем»" КАК Свойство
|ИЗ
| Справочник.Номенклатура КАК Номенклатура
|ГДЕ
| Номенклатура.Объем «»"»";

Результат = Запрос.Выполнить();

ВыборкаСсылка = Результат.Выгрузить();

Пока ВыборкаСсылка.Следующий() Цикл

ИсходящиеДанные = Новый Структура;
Ссылка = ВыборкаСсылка.Ссылка;
ИсходящиеДанные.Вставить(«ДополнительныеРеквизиты», Новый ТаблицаЗначений);

ДополнительныеРеквизиты = ИсходящиеДанные.ДополнительныеРеквизиты;
ДополнительныеРеквизиты.Колонки.Добавить(«Значение»);
ДополнительныеРеквизиты.Колонки.Добавить(«Свойство»);

Реквизит = ДополнительныеРеквизиты.Добавить();
Реквизит.Значение = ВыборкаСсылка.Объем;
Реквизит.Свойство = ВыборкаСсылка.Свойство;

ВыгрузитьПоПравилу(Ссылка, , ИсходящиеДанные, , «Свойства»);

КонецЦикла;
9) В пкс Свойства, в пкгс ДополнительныеРеквизиты для свойств указал ранее описанные правила.
10) Опробовал, работает.

progr-2008Ответить11.12.2012 в 18:05
Установила последние демо версии, последнюю тестовую 8.3 и переустановила для 8.3 библиотеку: regsvr32 «C:\Program Files\1cv8\8.3.2.163\bin\comcntr.dll»
Сначала разобралась с Дополнительными реквизитами в новых версиях, так как регистр сведений ЗначенияСвойствОбъектов в них больше не используется.
Включила возможность работы с дополниельными реквизитами и добавлила в БП дополнительные реквизиты Вес и Объем.
Так как использется Функциональная опция «Использовать дополнительные реквизиты и сведения» – включить в Администрирование, Настройка программы, закладка Дополнительные реквизиты и сведения, флаг Использовать дополнительные реквизиты и сведения. Администрирование, Формы – Наборы дополнительных реквизитов и сведения, выбрать Справочник Номенклатура. Подбор, Создать, Наименование – Вес, Тип значения – число, Записать и закрыть.
В Подбор, курсор Вес числовой, F9, отредактировать Наименование – Объем, Записать и закрыть. В Подбор Справочник Номенклатура – выбрать Вес и Объем, Записать и закрыть.
В карточке Номенклатуры внизу в Дополнительные реквизиты появились поля ввода числовых данных Вес и Объем. В ПВХ в Дополнительные реквизиты и сведения созданы два свойства – Вес и Объем.
Для теста – Номенклатура и Склад, Номенклатура, Создать Ном БП, Вес=123, Объем=456
Обращение к дополнительным реквизитам осуществляется как к табличной части справочника Номенклатуры, который соответствует объекту, Вес и Объем – в Свойство, числовые данные – в Значение.
СправочникНоменклатура ТабличнаяЧасть ДополнительныеРеквизиты
Свойство Тип ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения
ЗначениеТип Характеристика.ДополнительныеРеквизитыИСведения
Проверила в консоли запросов обращение к этой табличной части запросом.
ВЫБРАТЬ
НоменклатураДополнительныеРеквизиты.Ссылка как Объект,
НоменклатураДополнительныеРеквизиты.Свойство,
НоменклатураДополнительныеРеквизиты.Значение
ИЗ
Справочник.Номенклатура.ДополнительныеРеквизиты КАК НоменклатураДополнительныеРеквизиты
Получила результат: Объект Свойство Значение = Ном БП Вес 123 и Ном БП Объем 456

Настроила УТ по обмену с БП.
Обмен через прямое подключение к базе-корреспонденту, параметры:
Вариант работы информационной базы: Файловый, Каталог информационной базы: D:\БП30168Решение, Аутентификация операционной системы: Нет, Имя пользователя: Любимов (администратор)
Настройки обмена данными для этой информационной базы:
Документы выгружаются начиная с 01.01.2012; Отбор по организациям: Торговый дом «Комплексный»; Цены номенклатуры не выгружаются;
Подразделение по умолчанию: Бухгалтерия; Склад по умолчанию: Бытовая техника;
Для новых документов, загружаемых в эту базу, проведение разрешено;
Префикс информационной базы: ЦУ

Выполнила настройку обмена данными для БП:
Документы выгружаются начиная с 01.01.2012;
Выгружать для каждого контрагента отдельного партнера: да;
Номенклатурная группа по умолчанию для загружаемой номенклатуры: не указана;
Статья прочих доходов и расходов при оприходовании товаров: не указана;
Для новых документов, загружаемых в эту базу, проведение разрешено;
Префикс информационной базы: БП

Выполнила 2-х сторонний обмен.
Выгрузила из УТ правила обмена для редактировани.
Настройка, Изменить, Параметры обмена данными.
Типовые правила конфигурации, Выгрузить правила.
Правила обмена
Дата создания: 05.12.2012 12:48:31
Конфигурация источник: Управление торговлей, редакция 11.0 (11.0.9)
Конфигурация приемник: Бухгалтерия предприятия, редакция 3.0 (3.0.15)

Загрузила правила в Конвертацию.
Добавила два отдельных ПКО для ТЧ ДополнительныеРеквизиты для Номенклатуры с ПВД с произвольным алгоритмом.
Указать ПКО Номенклатура00001 .
Имя правила – ВесНоменклатуры
ВыборкаДанных = Новый Запрос;
ВыборкаДанных.Текст =
«ВЫБРАТЬ
| Номенклатура.Вес КАК Значение,
| Номенклатура.Ссылка КАК Объект,
| «»Вес»" КАК Свойство
|ИЗ
| Справочник. Номенклатура КАК Номенклатура
|ГДЕ
| Номенклатура.Вес 0;

Указать ПКО Номенклатура00002 .
Имя правила – ВесНоменклатуры
ВыборкаДанных = Новый Запрос;
ВыборкаДанных.Текст =
«ВЫБРАТЬ
| Номенклатура.Объем КАК Значение,
| Номенклатура.Ссылка КАК Объект,
| «»Объем»" КАК Свойство
|ИЗ
| Справочник. Номенклатура КАК Номенклатура
|ГДЕ
| Номенклатура.Объем 0;
Добавила ПКО для ПланВидовХарактеристикСсылка. ДополнительныеРеквизитыИСведения с поиском по Наименование.
Загрузила в УТ отредактированные правила обмена.
Настройка, Изменить, Параметры обмена данными.
Правила из файла, Загрузить правила, Каталог.
Выполнила обмен.

Игорь ИгнатовОтветить11.12.2012 в 17:02
Финальное задание.
1. Установил самые последние конфигурации УТ 11 и БП 3.0.
2. В БП создал дополнительные реквизиты Вес и Объем с типами как в УТ.
3. В конфигураторе УТ сохранил в файл правила обмена УТ с БП. Это макет ПравилаОбмена плана обмена ОбменУправлениеТорговлейБухгалтерияПредприятия30.
4. С помощью обработки MD82Exp.epf выгрузил в файлы метаданные УТ и БП.
5. В конфигурации Конвертация загрузил правила обмена УТ с БП. Затем обновил структуру метаданных УТ и БП (это необходимо, чтобы появились реквизиты Вес и Объем для создания ПКС).
6. Для ПКО Номенклатура создал 2 ПКС: Вес и Объем. Вес и объем будут передаваться в параметры объекта «Вес» и «Объем» соответственно. В соответсвующих обработчиках «Перед выгрузой» написал.

Значение = Источник.Вес;
Значение = Источник.Объем;
7. В обработчик после загрузки ПКО Номенклатура добавил следующий код:

Если НЕ Объект.ЭтоГруппа Тогда
    Если не ПараметрыОбъекта = Неопределено Тогда
        Сообщить("ПараметрыОбъекта[""Вес""]: " + ПараметрыОбъекта["Вес"]);
        Сообщить("ПараметрыОбъекта[""Объем""]: " + ПараметрыОбъекта["Объем"]);
        
        ХарактеристикаВес = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Вес");
        ХарактеристикаОбъем = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Объем");
        Если ЗначениеЗаполнено(ХарактеристикаВес) И ЗначениеЗаполнено(ХарактеристикаОбъем) Тогда
            СтрокаТЧ = Объект.ДополнительныеРеквизиты.Найти(ХарактеристикаВес, "Свойство");
            Если Не СтрокаТЧ = Неопределено Тогда 
                СтрокаТЧ.Значение = ПараметрыОбъекта["Вес"];     
            Иначе 
                СтрокаТЧ = Объект.ДополнительныеРеквизиты.Добавить();
                СтрокаТЧ.Свойство = ХарактеристикаВес;
                СтрокаТЧ.Значение = ПараметрыОбъекта["Вес"];
            КонецЕсли; 
            
            СтрокаТЧ = Объект.ДополнительныеРеквизиты.Найти(ХарактеристикаОбъем, "Свойство");
            Если Не СтрокаТЧ = Неопределено Тогда 
                СтрокаТЧ.Значение = ПараметрыОбъекта["Объем"];     
            Иначе 
                СтрокаТЧ = Объект.ДополнительныеРеквизиты.Добавить();
                СтрокаТЧ.Свойство = ХарактеристикаОбъем;
                СтрокаТЧ.Значение = ПараметрыОбъекта["Объем"];
            КонецЕсли; 
        КонецЕсли; 
    КонецЕсли; 
КонецЕсли;
8. Сохранил правила обмена в файл. Затем в текстовом редакторе сохранил текст правил в буфер обмена и вставил в макет плана обмена ОбменУправлениеТорговлейБухгалтерияПредприятия30. Данные правил загрузил в регистр.

9. Проверил работу онлайн обмена УТ-БП. Реквизиты Вес и объем из УТ успешно передаются в табличную часть ДополнительныеРеквизиты номенклатуры.

agonОтветить11.12.2012 в 15:28
Добрый день.
-Составление ментальной карты по важным (с Вашей точки зрения) моментам курса.
Это о чем и как?

Андрей ШнитовОтветить12.12.2012 в 05:08
Здравствуйте!

Кратко о том что такое ментальные карты можно прочитать здесь:
http://ru.wikipedia.org/wiki/%D0%94%D0%B8%D0%B0%D0%B3%D1%80%D0%B0%D0%BC%D0%BC%D0%B0_%D1%81%D0%B2%D1%8F%D0%B7%D0%B5%D0%B9
О программе Xmind:
http://ru.wikipedia.org/wiki/XMind
Здесь же есть ссылка на русскую версию.
Скачать программу Xmind (версия на английском языке) можно здесь:
http://www.xmind.net/
Краткая видео-инструкция к программе Xmind (для версии на русском языке):
https://www.youtube.com/watch?v=qhiglfxrJT4

Hunter1rОтветить11.12.2012 в 12:08
Настраиваем односторонний обмен из УТ11 в БП3.
1. Установил новые релизы УТ 11 и Бухгалтерия 3.0
2. Настроил On-line обмен между базами.
3. Выгрузил Метаданные из УТ11 и БП3
4. Сохранил типовые правила обмена из плана обмена «ОбменУправлениеТорговлейБухгалтерияПредприятия30″ из конфигурации УТ11.
Для успешного выполнения ДЗ необхоидмо было разобраться как работает механизм дополнительных реквизитов.
В ПВХ «Дополнительные реквизиты и сведения» необходимо создать два дополнительных реквизита «Вес» и «Объем». В БП установить флаг «Использовать дополнительные реквизиты и сведения». Администраирование\Настройки программы\Дополнительные реквизиты и сведения. При переносе данных необхоимдо заполнять справочник «Значения свойств и объектов». В котором указывается свойство и его значения. Кроме этого необходимо в правочник «Наборы дополнительных реквизитов и сведений» в предопределенный элемент «Справочник_Номенклатура» в ТЧ «ДополнительныеРеквизиты» добавить записи Вес и Объем. Таким образом система поймет что для номенклатуры используются набор дополнительных реквизитов Вес и Объем.
Модифицировал правила обмена. В ПКО Номенклатура в ПКС добавил ПКГС «ДополнительныеРеквизиты». Приемник указал «Значение» и «Свойство». У значения тип СправочникСсылка.ЗначенияСвойствОбъектов, а у Свойства тип ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения. Для них создал два ПКО. «ЗначенияСвойствОбъектов_ТребуетсяВладелец» – в которое передаю с помощью ВходящихДанных Владельца(свойство) и наименование(само значение). «ДополнительныеРеквизитыИСведения» – в которое передаю название свойства. Это делается для того, чтобы в БП в номенклатуре система могла заполнить доп. реквизиты «Значение» и «Свойство».
Чтобы правильно заполнились Вес и Объем в БП в ПКГС «ДополнительныеРеквизиты» описал произвольную коллекцию объектов:

ТЗ = новый ТаблицаЗначений;
ТЗ.Колонки.Добавить("Свойство");
ТЗ.Колонки.Добавить("Значение");
Стр= ТЗ.Добавить();
Стр.Свойство = "Вес";
Стр.Значение = Источник.Вес;
Стр= ТЗ.Добавить();
Стр.Свойство = "Объем";
Стр.Значение = Источник.Объем;
КоллекцияОбъектов = ТЗ;
На уровне Конвертации в обработчике «Перед загрузкой данных» написал код:

Пвх = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения;
    СвойствоВес = Пвх.НайтиПоНаименованию("Вес",Истина);
    Если СвойствоВес.Пустая() Тогда
        НовыйЭл = Пвх.СоздатьЭлемент();
        НовыйЭл.Наименование = "Вес";
        НовыйЭл.Записать();
        СвойствоВес = НовыйЭл.Ссылка;
    КонецЕсли;

    СвойствоОбъем = Пвх.НайтиПоНаименованию("Объем",Истина);
    Если СвойствоОбъем.Пустая() Тогда
        НовыйЭл = Пвх.СоздатьЭлемент();
        НовыйЭл.Наименование = "Объем";
        НовыйЭл.Записать();
        СвойствоОбъем = НовыйЭл.Ссылка;
    КонецЕсли;

    СвойстваНоменклатуры = Справочники.НаборыДополнительныхРеквизитовИСведений.Справочник_Номенклатура; 
    Если СвойстваНоменклатуры.ДополнительныеРеквизиты.Найти(СвойствоВес, "Свойство") = Неопределено Тогда
        ОбъектСвойстваНоменклатуры = СвойстваНоменклатуры.ПолучитьОбъект();
        НоваяСтрока = ОбъектСвойстваНоменклатуры.ДополнительныеРеквизиты.Добавить();
        НоваяСтрока.Свойство = СвойствоВес;
        //НоваяСтрока.ЗаполнятьОбязательно = 0;
        ОбъектСвойстваНоменклатуры.Записать();
    КонецЕсли;
    Если СвойстваНоменклатуры.ДополнительныеРеквизиты.Найти(СвойствоОбъем, "Свойство") = Неопределено Тогда
        ОбъектСвойстваНоменклатуры = СвойстваНоменклатуры.ПолучитьОбъект();
        НоваяСтрока = ОбъектСвойстваНоменклатуры.ДополнительныеРеквизиты.Добавить();
        НоваяСтрока.Свойство = СвойствоОбъем;
        //НоваяСтрока.ЗаполнятьОбязательно = 0;
        ОбъектСвойстваНоменклатуры.Записать();
    КонецЕсли;
При решении ДЗ с написанием этого кода возникали проблемы, но они разрешились:)
То есть перед загрузкой данных в БП создадутся записи в ПВХ Вес и Объем, а также в справочнике Наборы дополнительных реквизитов и сведений наборы Вес и Объем.
В результате выполнения переноса значения веса и объема перенесутся с помощью настроенного ПКО Номенклатура, а записи в ПВХ, в справочнике наборы доп. реквизитов и сведений при выполнении обработчика «Перед загрузкой данных».
Далее типовые правила обмена в УТ11 «ОбменУправлениеТорговлейБухгалтерияПредприятия30″ заменяем на модифицированные.
В результате выполнения ДЗ обеспечил односторонний перенос реквизитов «Вес» и «Объем» из УТ11 в БП3.

Alexey87Ответить11.12.2012 в 11:54
Создал две новые базу, УП и БП. Настрои между ними on line обмен с помощью стандартных средств. Далее изменил файл правил конвертации объектов. Добавил два параметра при выгрузке в БП («Объем» и «Вес»). После загрузки номенклатуры написал следующее:

//Подстановка параметров
Если ЗначениеЗаполнено(ПараметрыОбъекта["Вес"]) И  НЕ Объект.ЭтоГруппа Тогда
    Вес = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Вес");
    
    СтрокаТабличнойЧасти = Объект.ДополнительныеРеквизиты.Найти("Вес", "Объем");
    
    Если СтрокаТабличнойЧасти = Неопределено Тогда
        СтрокаТабличнойЧасти = Объект.ДополнительныеРеквизиты.Добавить();
    КонецЕсли;
    
    СтрокаТабличнойЧасти.Свойство = Вес;
    СтрокаТабличнойЧасти.Значение = Число(ПараметрыОбъекта["Вес"]);
КонецЕсли;

//Подстановка параметров
Если ЗначениеЗаполнено(ПараметрыОбъекта["Объем"]) И НЕ Объект.ЭтоГруппа Тогда
    Объем = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Объем");
    
    СтрокаТабличнойЧасти = Объект.ДополнительныеРеквизиты.Найти(Объем, "Свойство");
    
    Если СтрокаТабличнойЧасти = Неопределено Тогда
        СтрокаТабличнойЧасти = Объект.ДополнительныеРеквизиты.Добавить();
    КонецЕсли;
    
    СтрокаТабличнойЧасти.Свойство = Объем;
    СтрокаТабличнойЧасти.Значение = Число(ПараметрыОбъекта["Объем"]);
КонецЕсли;
.
В БП создал 2 дополнительных значения в ПВХ(«Вес», «Объем»). Настроил их тип значения и указал что входят в группу номенклатуры. Проверил перенос доп. реквизитов.

Краснокутская СветланаОтветить11.12.2012 в 11:25
Обновила свои БП 3.0 и УТ 11 до последних версий. Выгрузила типовые правила конвертации. Внесла изменение в ПКО Номенклатура: добавила ПКГС дополнительные реквизиты. В обработчике ПередОбработкой:

 Если Не Источник.ЭтоГруппа Тогда
    КоллекцияОбъектов = Новый ТаблицаЗначений;
    КоллекцияОбъектов.Колонки.Добавить("Значение");
    КоллекцияОбъектов.Колонки.Добавить("Свойство");
    КоллекцияОбъектов.Колонки.Добавить("ТекстоваяСтрока");
    
    
    Вес = КоллекцияОбъектов.Добавить();
    Вес.Значение = Источник.Вес;
    Вес.ТекстоваяСтрока = "Вес";
    Вес.Свойство = "Вес";
    
    Объем = КоллекцияОбъектов.Добавить();
    Объем.Значение = Источник.Объем;
    Объем.ТекстоваяСтрока = "Объем";
    Объем.Свойство = "Объем";
иначе
    отказ = истина;
КонецЕсли;
Создала ПКО СвойстваНоменклатура. Добавила ПКС Наименование, по нему же установила Поиск. Поиск по УИД отключила.
В БП 3.0 создала для элемента Номенклатура справочника Набор доп. реквизитов два новый реквизита: Вес и Объем.
В УТ и БП создала план обмена. в УТ указала для использования новые правила. Обмен прошел успешно.
Очень хотелось заполнить Набор дополнительных реквизитов при загрузке, но понимаю что не успеваю, поэтому слегка упростила задачу

14. [image: image15.png]



allzonОтветить11.12.2012 в 11:07
1) Включил возможность вести дополнительные реквизиты в БП 3.0 – «Администрирование» – «Настройки програмы» – «Дополнительные реквизиты» – «Использовать доп. реквизиты».
2) Добавил «Вес» и «Объем» в доп. реквизиты справочника «Номенклатура» в БП 3.0 .
3) В конфигураторе УТ 11 сохранил правила обмена УТ11-БП30 и загрузил в КД.
4) В правилах создал пко Свойства для справочника «Номенклатура»(с пустым источником) добавил пкгс для ТЧ «ДополнительныеРеквизиты» и установил галки получения из входящих данных.
5) Создал пко ДопРеквизитыИСведения (пвх), с пкс наименование и ПередВыгрузкой написал:

    Значение = Источник;
6) Создал пко ЗначенияСвойствОбъектовМоё (справочник) с пкс Наименование и ПередВыгрузкой написал:

    Значение = Источник;
Для пкс Владелец указал значение, в зависимости от свойства.

7) Создал пвд Вес с произвольным алгоритмом, с пко Свойства и перед обработкой написал:

    Запрос = Новый Запрос;
    Запрос.Текст = 
        "ВЫБРАТЬ
        |   Номенклатура.Ссылка,
        |   Номенклатура.Вес,
        |   ""Вес"" КАК Свойство
        |ИЗ
        |   Справочник.Номенклатура КАК Номенклатура
        |ГДЕ
        |   Номенклатура.Вес <> """"";

    Результат = Запрос.Выполнить();

    ВыборкаСсылка = Результат.Выгрузить();

    Пока ВыборкаСсылка.Следующий() Цикл
        
        ИсходящиеДанные = Новый Структура;
        Ссылка = ВыборкаСсылка.Ссылка;
        ИсходящиеДанные.Вставить("ДополнительныеРеквизиты", Новый ТаблицаЗначений);

        ДополнительныеРеквизиты = ИсходящиеДанные.ДополнительныеРеквизиты;
        ДополнительныеРеквизиты.Колонки.Добавить("Значение");
        ДополнительныеРеквизиты.Колонки.Добавить("Свойство");
                
        Реквизит = ДополнительныеРеквизиты.Добавить();
        Реквизит.Значение = ВыборкаСсылка.Вес;
        Реквизит.Свойство = ВыборкаСсылка.Свойство;    
        
        ВыгрузитьПоПравилу(Ссылка, , ИсходящиеДанные, , "Свойства");
        
    КонецЦикла;
8) Создал пвд Объем с произвольным алгоритмом, с пко Свойства и перед обработкой написал:

    Запрос = Новый Запрос;
    Запрос.Текст = 
        "ВЫБРАТЬ
        |   Номенклатура.Ссылка,
        |   Номенклатура.Объем,
        |   ""Объем"" КАК Свойство
        |ИЗ
        |   Справочник.Номенклатура КАК Номенклатура
        |ГДЕ
        |   Номенклатура.Объем <> """"";

    Результат = Запрос.Выполнить();

    ВыборкаСсылка = Результат.Выгрузить();

    Пока ВыборкаСсылка.Следующий() Цикл
        
        ИсходящиеДанные = Новый Структура;
        Ссылка = ВыборкаСсылка.Ссылка;
        ИсходящиеДанные.Вставить("ДополнительныеРеквизиты", Новый ТаблицаЗначений);

        ДополнительныеРеквизиты = ИсходящиеДанные.ДополнительныеРеквизиты;
        ДополнительныеРеквизиты.Колонки.Добавить("Значение");
        ДополнительныеРеквизиты.Колонки.Добавить("Свойство");
                
        Реквизит = ДополнительныеРеквизиты.Добавить();
        Реквизит.Значение = ВыборкаСсылка.Объем;
        Реквизит.Свойство = ВыборкаСсылка.Свойство;    
        
        ВыгрузитьПоПравилу(Ссылка, , ИсходящиеДанные, , "Свойства");
        
    КонецЦикла;
9) В пкс Свойства, в пкгс ДополнительныеРеквизиты для свойств указал ранее описанные правила.
10) Опробовал, работает.

1. Были модифицированны типовые правила обмена УТ 11.0 -> БП 3.0:
1) Добавлено ПКО -> ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения с наименованием «ДополнительныеРеквизитыИСведенияВесИОбъем». В обработчике ПриВыгрузке ПКО добавлен код:

Если ТипЗнч(ВходящиеДанные) = Тип("Структура") Тогда
    ТипЗначения = "";
    ВходящиеДанные.Свойство("ТипЗначения", ТипЗначения);
    Если ТипЗначения <> Неопределено Тогда
        УзелТипы = СоздатьУзел("Типы");
        УзелТип = СоздатьУзел("Тип");
        УзелТип.ЗаписатьТекст(ТипЗначения);
        ДобавитьПодчиненный(УзелТипы, УзелТип);
        ДобавитьПодчиненный(Приемник, УзелТипы);
    КонецЕсли;
КонецЕсли;
Отключен поиск по уникальному идентификатору, включена автоматическая генерация кода. Создано ПКС «Наименование», в обработчике ПередВыгрузкой добавлен код:

Значение = Источник;
Создано ПКС «ЭтоДополнительноеСведение», в обработчике ПередВыгрузкой добавлен код:

Значение = Ложь;
Включен поиск по свойству «Наименование».
2) Модифицировано ПКО «Номенклатура». Добавлено ПКГС -> ДополнительныеРеквизиты. В обработчик ПередВыгрузкой добавлен код:

Если Источник.ЭтоГруппа Тогда
    Отказ = Истина;
КонецЕсли;
КоллекцияОбъектов  = Новый ТаблицаЗначений;
КоллекцияОбъектов.Колонки.Добавить("Значение");
КоллекцияОбъектов.Колонки.Добавить("Свойство");
Если (Источник.Вес <> 0  ИЛИ Источник.Объем <> 0) Тогда
    Если Источник.Вес <> 0  Тогда
        СтрокаДопРеквизитов = КоллекцияОбъектов.Добавить();
        СтрокаДопРеквизитов.Значение = Источник.Вес;
        СтрокаДопРеквизитов.Свойство = "Вес";
    КонецЕсли;
    Если Источник.Объем <> 0  Тогда
        СтрокаДопРеквизитов = КоллекцияОбъектов.Добавить();
        СтрокаДопРеквизитов.Значение = Источник.Объем;
        СтрокаДопРеквизитов.Свойство = "Объем";
    КонецЕсли;
КонецЕсли;
Добавлены ПКС «Значение» и «Свойство». Для ПКС «Свойство» указано ПКО «ДополнительныеРеквизитыИСведенияВесИОбъем». В обработчике ПередВыгрузкой ПКС «Свойство» добавлен код:

ИсходящиеДанные = Новый Структура;
Если ОбъектКоллекции.Свойство =  "Вес"  ИЛИ  ОбъектКоллекции.Свойство =  "Объем"  Тогда
    ИсходящиеДанные.Вставить("ТипЗначения", "Число");
КонецЕсли;
2. В пользовательском режиме в конфигурации БП 3.0 включена функциональная опция использования обмена данными, установлен префикс ИБ «БП». С помощью «Помощника создания обмена данными» настроены параметры обмена: выбран каталог обмена информацией, заданы наименования и префикс ИБ, установлены ограничения передачи данных и значения по умолчанию.
В конфигурации УТ 11.0 создан новый обмен с Бухгалтерией 3.0. В «Помощнике создания обмена данными» выбран вариант «Продолжить создание обмена данными» и выбран файл настроек, выгруженный из БП 3.0, установлены ограничения передачи данных и значения по умолчанию.
В УТ 11.0 загружены модифицированные правила обмена, созданные в п.1. Произведена выгрузка и загрузка данных. Информация о весе и объеме номенклатуры переносятся в ТЧ справочника «Дополнительные реквизиты» корректно.

В БП 3.0 Администрирование – настройка программы – Дополнительные реквизиты: устанавливаем флажок использования. В наборах дополнительных реквизитов и сведений, для справочника Номенклатура добавляем два реквизита типа Число: Вес и Объем. Настраиваем обмен УТ-БП. Выгружаем правила конвертации объектов. Загружаем правила в конфигурацию конвертация данных.
создаем ПКО Дополнительные сведения. Объект-приемник РегистрСведенийЗапись.ДополнительныеСведения. создаем ПКО ДопРеквНоменклатуры: приемник ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения.
В ПКО ДополнительныеСведения создаем ПКС Объект, свойство, Значение. Для ПКС Свойство устанавливаем правило ДополнительныеРеквизитыИСведения. Перед выгрузкой помещаем код

ИсходящиеДанные = Новый Структура;
Если Источник.Свойство = «Вес» Тогда
ИсходящиеДанные.Вставить(«ТипЗначения», «Число»);
ИначеЕсли Источник.Свойство = «Объем» Тогда
ИсходящиеДанные.Вставить(«ТипЗначения», «Число»);
КонецЕсли;

Для ПКС Значение Перед выгрузкоЙ помещаем код: Значение = Источник.
создаем ПВД Вес. Способ выборки Произвольный алгоритм. ПК ДополнительныеСведения. Перед выгрузкой:

ВыборкаДанных = Новый Запрос;
ВыборкаДанных.Текст =
«ВЫБРАТЬ
| Номенклатура.Ссылка КАК Объект,
| Номенклатура.Вес КАК Значение,
| «»Вес»" КАК Свойство
|ИЗ
| Справочник.Номенклатура КАК Номенклатура
|ГДЕ
| Номенклатура.Вес < 0";

Аналогично создаем ПВД Объем.
сохраняем правила. Загружаем в УТ в режиме Правила из файла.

1. Для выполнения финального задания, загрузим последние версии УТ 11 БП 3.0
2. Настроим обмен между ними.
3. Выгрузим правила конвертации объектов из УТ.
4. Для начала нам нужно настроить БП 3.0.
5. В настройках включим использование дополнительных реквизитов и сведений.
6. В подсистеме администрирование в панели навигации выбирем – Формы – Наборы доп-х реквизитов и сведений.
7. Найдем там справочник номенклатура и создадим там два реквизита Вес и Объем. Настроим их формат такой же как в УТ.
8. Настроим форму элемента номенклатуры. Все действия – изменить форму. Включим отображение ТЧ Дополнительные реквизиты. Все.
9. Идем в КД. Создаем новое ПКГС для ТЧ Дополнительные реквизиты и в обработчике перед обработкой пишем код:

КоллекцияОбъектов = Новый ТаблицаЗначений;
КоллекцияОбъектов.Колонки.Добавить("Свойство");
КоллекцияОбъектов.Колонки.Добавить("Значение");

Если ЗначениеЗаполнено(Источник.Вес) Тогда
    НС = КоллекцияОбъектов.Добавить();   
    НС.Свойство = "Вес";
    НС.Значение = Источник.Вес;
КонецЕсли;

Если ЗначениеЗаполнено(Источник.Объем) Тогда
    НС = КоллекцияОбъектов.Добавить();   
    НС.Свойство = "Объем";
    НС.Значение = Источник.Объем;
КонецЕсли;
10. Свойство является ссылочным типом и для него создадим ПКО ПланВидовХарактеристик.ДополнительныеРеквизитыИСведенияПеренос. Выключим поиск по УИД и оставим одно свойство – Наименование, куда будем передавать строковое значение. И присвоим его значение:

Значение = Источник;
11. Установим принудительно использование данного ПКО для реквизита Свойство.
12. Так как значение имеет составной тип нужно передать тип, в нашем случае это всегда число:

ТипПриемника = "Число";
Все.
13. Сохраним правила. Загрузим их в правила конвертации объектов нашего обмена.
14. Изменим любой элемент, установив вес и объем. Сделаем выгрузку данных.
15. Сделаем загрузку данных и проверим значения. Все отрабатываем корректно.
16. Финальное задание успешно выполнено!!!

////////////////////////////////
// РЕШЕНИЕ ПРАКТИЧЕСКОГО ЗАДАНИЯ

Сначала начал копаться в РС Дополнительные сведения,
но потом внимательнее прочитал задание :)

Оказалось всё значительно проще. Идея следующая, для ПКО Номенклатура создаём ПКГС,
и программно заполняем две строки.

В ПКО для ПВХ передаём все поля через структуру.
Оригинальное ПКО для ПВХ использовать не получилось. Ошибка:

{Обработка.КонвертацияОбъектовИнформационныхБаз.МодульОбъекта(9411)}: Метод объекта не обнаружен (УникальныйИдентификатор)
УникальныйИдентификатор = СсылкаНаИсточник.УникальныйИдентификатор();
Так что создал своё. С переносом нескольких свойств и ТипЗначения.

Код формирования таблицы для ПКГС:

СвойствоВес = Новый Структура;
СвойствоВес.Вставить("Наименование", "Вес");
СвойствоВес.Вставить("ЭтоГруппа", Ложь);

СвойствоОбъем = Новый Структура;
СвойствоОбъем.Вставить("Наименование", "Объем");
СвойствоОбъем.Вставить("ЭтоГруппа", Ложь);

КоллекцияОбъектов = Новый ТаблицаЗначений;
КоллекцияОбъектов.Колонки.Добавить("Значение");
КоллекцияОбъектов.Колонки.Добавить("Свойство");
КоллекцияОбъектов.Колонки.Добавить("ТекстоваяСтрока");

Строка = КоллекцияОбъектов.Добавить();
Строка.Свойство = СвойствоВес;
Строка.Значение = Источник.Вес;
Строка.ТекстоваяСтрока = "Загружено из УТ 11";

Строка = КоллекцияОбъектов.Добавить();
Строка.Свойство = СвойствоОбъем;
Строка.Значение = Источник.Объем;
Строка.ТекстоваяСтрока = "Загружено из УТ 11";
Дополнительные свойства, я так понял смотреть по кнопке, рядом с сохранить.
Вновь загруженные свойства туда не попадают, т.к. используется всё же РС Дополнительные сведения. :)
Однако, запрос к табличной части номенклатуры показывает, что свойства загрузились правильно.

////////////////////////////////
// МЫСЛЕКАРТА

http://zalil.ru/34054241
У меня получилась скорее иерархия элементов обмена
с указанием важных, на мой взгляд, свойств и обработчиков.

Может кому и пригодится. :)

Программа только не порадовала, пару часов бился чтобы заработала
потом ещё и глючила, в общем неудовольствие от использования.

////////////////////////////////
// ОТЗЫВ

http://mg.spec8.ru/2012/10/26/feedback/comment-page-2/#comment-67227

TregardОтветить10.12.2012 в 17:04
Выполнил.
В БП включил использование дополнительных реквизитов и свойств.
В обработчике «Перед загрузкой данных» правил конвертации добавил для справочника Номенклатуры Объём и Вес:

Объём = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Объём");
Если НЕ ЗначениеЗаполнено(Объём) Тогда
    Объём = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.СоздатьЭлемент();
    Объём.Наименование = "Объём";
    ТипЧисло = ОбщегоНазначенияБПКлиентСервер.ПолучитьОписаниеТиповЧисла(15, 3, ДопустимыйЗнак.Неотрицательный);
    Объём.ТипЗначения = ТипЧисло;
    Объём.Записать();
    Объём = Объём.Ссылка;
КонецЕсли;
Вес = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Вес");
Если НЕ ЗначениеЗаполнено(Вес) Тогда
    Вес = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.СоздатьЭлемент();
    Вес.Наименование = "Вес";
    ТипЧисло = ОбщегоНазначенияБПКлиентСервер.ПолучитьОписаниеТиповЧисла(15, 3, ДопустимыйЗнак.Неотрицательный);
    Вес.ТипЗначения = ТипЧисло;
    Вес.Записать();
    Вес = Вес.Ссылка;
КонецЕсли;
ДополнительныеРеквизитыНоменклатуры = Справочники.НаборыДополнительныхРеквизитовИСведений.Справочник_Номенклатура.ПолучитьОбъект();
Если ДополнительныеРеквизитыНоменклатуры.ДополнительныеРеквизиты.Найти(Объём,"Свойство") = Неопределено Тогда
    НовыйРеквизит = ДополнительныеРеквизитыНоменклатуры.ДополнительныеРеквизиты.Добавить();
    НовыйРеквизит.Свойство = Объём;
КонецЕсли;
Если ДополнительныеРеквизитыНоменклатуры.ДополнительныеРеквизиты.Найти(Вес,"Свойство") = Неопределено Тогда
    НовыйРеквизит = ДополнительныеРеквизитыНоменклатуры.ДополнительныеРеквизиты.Добавить();
    НовыйРеквизит.Свойство = Вес;
КонецЕсли;
Если ДополнительныеРеквизитыНоменклатуры.Модифицированность() Тогда
    ДополнительныеРеквизитыНоменклатуры.Записать();
КонецЕсли;
Обновил структуру метаданных УТ. Настроил перенос Веса и Объёма в параметры.
«После загрузки» ПКО Номенклатура заполняю значения веса и объёма в табличной части ДополнительныеРеквизиты:

Объём = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Объём");
Вес = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Вес");

Если НЕ Объект.ЭтоГруппа Тогда
    НовыйРеквизитНоменклатуры = Объект.ДополнительныеРеквизиты.Добавить();
    НовыйРеквизитНоменклатуры.Значение = ПараметрыОбъекта["Объем"];
    НовыйРеквизитНоменклатуры.Свойство = Объём;
    
    НовыйРеквизитНоменклатуры = Объект.ДополнительныеРеквизиты.Добавить();
    НовыйРеквизитНоменклатуры.Значение = ПараметрыОбъекта["Вес"];
    НовыйРеквизитНоменклатуры.Свойство = Вес;
КонецЕсли;

НатальяОтветить10.12.2012 в 13:13
Загрузила в Конвертацию данных типовые правила конвертации объектов Плана обмена «ОбменУправлениеТорговлейБухгалтерияПредприятия30».
Добавила новое служебное ПКО «ВГХ» для ПВХ «ДополнительныеРеквизитыИСведения». Отключила поиск по внутреннему идентификатору. Включила автогенерацию кода. В обработчике «После загрузки»:

Если Не ОбъектНайден Тогда
Объект.ТипЗначения = Новый ОписаниеТипов("Число");
КонецЕсли
Добавила ПКС:
«Наименование», установила флажок поиска по свойству. В обработчике «Перед выгрузкой»:

Значение = Источник

«ЭтоДополнительноеСведение» – В обработчике «Перед выгрузкой»:

  Значение = Ложь;
В ПКО «Номенклатура» добавила ПКГС «ДополнительныеРеквизиты» с пустым источником. Установила флажок «Получить из входящих данных». В обработчике «Перед выгрузкой»:

Отказ = Источник.ЭтоГруппа;
Если Не Отказ Тогда
ДопРеквизиты = Новый ТаблицаЗначений;
ДопРеквизиты.Колонки.Добавить("Значение");
ДопРеквизиты.Колонки.Добавить("Свойство");
НоваяСтрока = ДопРеквизиты.Добавить();
НоваяСтрока.Свойство = "Вес";
НоваяСтрока.Значение = Источник.Вес;
НоваяСтрока = ДопРеквизиты.Добавить();
НоваяСтрока.Свойство = "Объем";
НоваяСтрока.Значение = Источник.Объем;
ВходящиеДанные = Новый Структура;
ВходящиеДанные.Вставить("ДополнительныеРеквизиты",ДопРеквизиты);
КонецЕсли;
В ПКГС «ДополнительныеРеквизиты» добавила ПКС с установленными флажочками «Получить из входящих данных»:
«Значение» – в обработчике «перед выгрузкой»:

Значение = ОбъектКоллекции.Значение;
«Свойство» – с правилом конвертации «ВГХ» в обработчике «перед выгрузкой»:

Значение = ОбъектКоллекции.Свойство;
В настройках бухгалтерии установила флажок «Использовать дополнительные реквизиты и сведения».
Загрузила измененные правила обмена в макет плана обмена «Правила обмена».
После обмена данными в бухгалтерии в настройках дополнительных реквизитов для справочника номенклатура добавила загруженные дополнительные реквизиты «Вес» и «Объем».

vlasОтветить10.12.2012 в 11:36
Здравствуйте!
Практическая задача выполнена так:
- Установлены типовые УТ (11.0.9.12) и БП (3.0.15.5).
- Настроены узлы плана обмена в обеих конфигурациях.
- Из УТ выгружены типовые правила обмена с БП.
- В правила обмена внесены следующие изменения:
Для Номенклатура добавлено ПКГС ДополнительныеРеквизиты с пустым источником. В ПКГС формируется Таблица значений с колонками Значение и Свойство (Отказ = ЭтоГруппа). В ТЗ добавляются 2-е строки – Объекм и Вес в качестве своиства, значения берутся из источника. В ПКС для реквизита Свойство создано и назначено служебное ПКО РеквизитыНоменклатуры (приемник ПВХ.ДополнительныеРеквизитыИСведения, поиск по Наименованию, ЭтоГруппа, ЭтоДополнительноеСведение).
- В БП в ПВХ Дополнительные реквизиты и сведения добавлены два свойства Вес и Объем (тип Число), в справочник Наборы дополнительных реквизитов и сведений (элемент Справочник «Номенклатура») в ТЧ Дополнительные реквизиты добавлены созданные на предыдущем шаге свойства.
- Обмен работает нормально.
Это была Лайт версия модифицированных правил. В полной версии предполагалось программное создание в ПВХ дополнительных реквизитов и добавление из в ТЧ с помощью правил обмена. Но неожиданно столкнулся с проблемой – для новых свойств Вес и Объем не удавалось задать тип (в моем случае Число), новые свойства всегда создавались с типом по-умолчанию. Так же новые свойства не добавлялись в ТЧ справочника Наборы дополнительных реквизитов и сведений (предполагалось получение объекта из этого справочника, модификация ТЧ и запись объекта).
Сам механизм был подсмотрен в типовых правилах (справочник ДоговорыКонтрагентов, только там добавлялись Свойства, а не реквизиты). Весь алгоритм был создан и протестирован в обработке Универсальный обмен XML – он работал корректно. При переносе же в онлайн обмен, дополнительный узел с заданием типа элемента ПВХ не формировался (код в событиях ПриВыгрузке/ПослеВыгрузки), ТЧ справочника не обновлялась и запись объекта не происходила. Все было похоже на то, что код из обработчиков не выполняется, хотя правила были созданы в режиме совместимости. Перевести правила в режим с отключенной совместимостью не удалось, т.к. в УТ интегрирована старая версия БСП и в ней отсутствуют механизмы для обмена с отключенной совместимостью. КД использовалась версии 1.6.4 и 1.6.5. Проблему так и не удалось решить, поэтому в качестве решения представлена версия Лайт;):) Но тем не менее, в этом есть плюс – на попытки решить проблему ушло 3 дня, за которые вдоль и поперек изучил структуру XML файла обмена:) Если до этого проще было сделать обмен, чтобы проверить правила, то теперь проще открыть файл обмена в браузере:):)

dumalОтветить10.12.2012 в 10:14
Поправка. Удалось решить задачу и первым вариантом, добавив служебное ПКО по переносу ПВХ

Сергей ТарановОтветить10.12.2012 в 07:41
1 Создаем обмен Торговля 11 – Бухгалтерия 3.0
Тестируем—Обмен работает
2 Создаем дополниткльные реквизиты в бухгалтерии 3.0
ВЕС – число Объем – число
3 Выглужаем правилп обмена УТ-Бух в файл
4 Загружаем в Конвертацию для редактирования
Обнаруживаем, что в типовых правилах отсутствует реквизит «Объем»
Стандартным способом обновляем метаданные УТ11 в Конвертации
Донастраиваем ПКО «Номенклатура»
Добавляем табличнуючасть.ДополнительныеРеквизиты.
Определяем обработчик для ПКГС

КоллекцияОбъектов = Новый ТаблицаЗначений;
КоллекцияОбъектов.Колонки.Добавить("Свойство");
КоллекцияОбъектов.Колонки.Добавить("Значение");
Стр =  КоллекцияОбъектов.Добавить();
Стр.Свойство = "Вес";
Стр.Значение = Источник.Вес;
Стр =  КоллекцияОбъектов.Добавить();
Стр.Свойство = "Объем";
Стр.Значение = Источник.Объем;
Определяем два ПКС Значение и Свойство ,
Для которых создаем ПКО «ДополнительныеРеквизитыИСведения_1″
Синхронизацию определяем но наименованию и доопределяем ПКС «ПередВыгрузкой»,

Значение = Источник;
Загружаем новые правила. Тестируем, Все переносится в табличнуючасть.ДополнительныеРеквизиты.

РамильОтветить09.12.2012 в 23:04
Задание делим на два этапа.

1. этап

Создаем вспомогательное ПКО для плана видов характеристик «Дополнительные реквизиты и сведения» с пустым источником.
Синхронизацию по УИД снимаем, автозаполнение кода не ставим потому что длина кода нулевая. В обработчике «Перед выгрузкой» пишем

КлючВыгружаемыхДанных = "ДополнительныеРеквизитыИСведения"+Источник;
ЗапоминатьВыгруженные = Истина;
В обработчике «Поля поиска» пишем

// ищем по наименованию
НаименованиеПоиска = СвойстваПоиска["Наименование"];
СсылкаПВХ = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию(НаименованиеПоиска,Истина);

Если НЕ ЗначениеЗаполнено(СсылкаПВХ) Тогда
    
    // если не найден, то создаем новый и заполняем тип значения
    НовыйОбъектПВХ = ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.СоздатьЭлемент();
    НовыйОбъектПВХ.Наименование = НаименованиеПоиска;
    ТипОбъектаПВХ = Новый ОписаниеТипов("Число",,,Новый КвалификаторыЧисла(15,3));
    НовыйОбъектПВХ.ТипЗначения = ТипОбъектаПВХ;
    НовыйОбъектПВХ.ФорматСвойства = "ЧДЦ='3'";
    Попытка
    НовыйОбъектПВХ.ОбменДанными.Загрузка = Истина;
    Исключение
    КонецПопытки;
    НовыйОбъектПВХ.Записать();
    СсылкаПВХ = НовыйОбъектПВХ.Ссылка;
    
    // добавим новую характеристику в набор реквизитов
    СпрОбъект = Справочники.НаборыДополнительныхРеквизитовИСведений.Справочник_Номенклатура.ПолучитьОбъект();
    НСтр = СпрОбъект.ДополнительныеРеквизиты.Добавить();
    НСТр.Свойство = СсылкаПВХ;
    СпрОбъект.Записать();
    
КонецЕсли;

СсылкаНаОбъект = СсылкаПВХ;
Добавляем ПКС «Наименование» с флагом поиска. В обработчике ПКС «Перед выгрузкой» пишем

Значение = Источник;
2. этап

Для ПКО «Номенклатура» создаем новый ПКГС «ДополнительныеРеквизиты» с пустым источником
и двумя ПКС «Свойство» и «Значение». Для ПКС «Свойство» принудительно указываем созданное нами вспомогательное правило конвертации.
Для ПКГС в обработчике «Перед обработкой» пишем код

Если Источник.ЭтоГруппа Тогда
    Отказ = Истина;
КонецЕсли;

Если ЗначениеЗаполнено(Источник.Вес) ИЛИ ЗначениеЗаполнено(Источник.Объем) Тогда

    КоллекцияОбъектов = Новый ТаблицаЗначений;
    КоллекцияОбъектов.Колонки.Добавить("Значение");
    КоллекцияОбъектов.Колонки.Добавить("Свойство");

    Если ЗначениеЗаполнено(Источник.Вес) Тогда
        НСтр = КоллекцияОбъектов.Добавить();
        НСтр.Свойство = "Вес";
        НСтр.Значение = Источник.Вес;
    КонецЕсли;

    Если ЗначениеЗаполнено(Источник.Объем) Тогда
        НСтр = КоллекцияОбъектов.Добавить();
        НСтр.Свойство = "Объем";
        НСтр.Значение = Источник.Объем;
    КонецЕсли;

КонецЕсли;

tduОтветить09.12.2012 в 19:36
Финальное задание:
1) В базе БП 3.0 в настройках программы включил опцию «Использовать дополнительные реквизиты и сведения», чтобы был доступен механизм дополнительных реквизитов, в пункте «Наборы дополнительных реквизитов и свойств» для справочника «Номенклатура» указал два дополнительных числовых реквизита «Вес» и «Объем»;
2) Настроил онлайн-обмен со стандартными правилами, перенос выполняется успешно;
3) Выгрузил правила УТ11 – БП3.0, загрузил их в базу конвертации данных, из баз УТ 11 и БП 3.0 выгрузил описание структуры метаданных и загрузил их с модификацией конфигураций в базу конвертации;
4) В ПКО «Номенклатура» добавил ПКГС «ДополнительныеРеквизиты» внутри создал 3 ПКС без источника: Значение, Свойство, ТекстоваяСтрока;
5) Для поиска ПВХ по наименованию добавил ПКО «ДополнительныеРеквизитыИСведенияПоНаименованию» без источника, автоматическая генерация кодов, 1 ПКС «Наименование» включил по нему поиск, в обработчике «Перед выгрузкой» установил Значение = Источник;
6) В ПКО «Номенклатура» в ПКГС «ДополнительныеРеквизиты» для ПКС «Свойство» указал правило конвертации «ДополнительныеРеквизитыИСведенияПоНаименованию»;
7) В ПКГС «ДополнительныеРеквизиты» в обработчике «Перед Обработкой» определяю коллекцию объектов, для этого создал таблицу значений с колонками: Значение, Свойство, ТекстоваяСтрока и если вес более нуля, тогда добавляю в таблицу значений строку c заполненным весом и свойством «Вес», аналогично для объема.

КоллекцияОбъектов = Новый ТаблицаЗначений;
КоллекцияОбъектов.Колонки.Добавить("Значение");
КоллекцияОбъектов.Колонки.Добавить("Свойство");
КоллекцияОбъектов.Колонки.Добавить("ТекстоваяСтрока");
Если Источник.Вес > 0 Тогда // заполнен Вес
    НоваяСтрока = КоллекцияОбъектов.Добавить();
    НоваяСтрока.Значение = Источник.Вес;
    НоваяСтрока.Свойство = "Вес";
    НоваяСтрока.ТекстоваяСтрока = "";
КонецЕсли;
Если Источник.Объем > 0 Тогда // заполнен Объем
    НоваяСтрока = КоллекцияОбъектов.Добавить();
    НоваяСтрока.Значение = Источник.Объем;
    НоваяСтрока.Свойство = "Объем";
    НоваяСтрока.ТекстоваяСтрока = "";
КонецЕсли;
8) Сохранил правила в файл, в базе УТ 11 в пункте «Обмен данными» в созданном обмене «УТ 11 – БП3.0″ в настройках в Параметрах обмена данными загрузил правила конвертации объектов из файла;
9) Для нескольких элементов справочника номенклатуры в базе УТ 11 заполнил «Вес» и «Объем»;
10) Выполнил обмен данными, он прошел успешно, в базе БП 3.0 открыл элементы, для которых ранее указал «Вес» и «Объем», значения перенеслись правильно;
11) В конфигураторе базы УТ 11 включил возможность изменения и для плана обмена «ОбменУправлениеТорговлейБухгалтерияПредприятия30″ в макет ПравилаОбмена записал отредактированные мною правила конвертации объектов, запустил 1С в пользовательском режиме и в параметрах обмена данных загрузил правила из конфигурации.

ulanОтветить09.12.2012 в 18:20
Сначала включил в настройках программы БП использоване Дополнительных сведений и реквизитов.
Взял типовые правила обмена из шаблона установки УТ и загрузил их в КД.
Для ПКО Номенклатура добавил ПКГС ДополнительныеРеквизиты в качестве источника ничего не указывал, а формировал Коллекцию объектов в обработчике события Перед обработкой в зависимости от значений реквизитов Вес и Объем.
Создал служебное ПКО для конвертации реквизитов Вес и Объем в ПВХ ДополнительныеРеквизитыИСведения, зная об проблемах с типом значения, передавал его черех параметр и устанавливал в обработчике После загрузки.
В этом же обработчике для новых объектов добавлял для предопределенного элемента справочника «Наборы дополнительных реквизитов и сведений» в табличную часть «Дополнительные реквизиты» созданные из реквизитов Вес и Объем ссылки ПВХ ДополнительныеРеквизитыИСведения.
Вроде все просто, но отладка заняла достаточно много времени поскольку для того, чтобы полноценно протестировать правила через УОД пришлось вносить различные изменения в них, связанные с комментированием участков кода работающих при on-line обмене и вызывающих ошибки при использовании УОД.

LaurusОтветить09.12.2012 в 18:06
1. Устанавливаем последние версии УТ11 и БП3. Загружаем в конвертацию последнюю версию типовых правил, а затем обновляем описание конфигураций.
2. Для ПКО «Номенклатура» добавляем ПКГС «ДополнительныеРеквизиты» и принадлежащие ему ПКС (все с пустым источником). В обработчике Перед обработкой ПКГС добавляем код:

Если Не Источник.ЭтоГруппа Тогда
    КоллекцияОбъектов = Новый ТаблицаЗначений;
    КоллекцияОбъектов.Колонки.Добавить("Значение");
    КоллекцияОбъектов.Колонки.Добавить("Свойство");
    КоллекцияОбъектов.Колонки.Добавить("ТекстоваяСтрока");

    // вес номенклатуры
    СтрокаКоллекции = КоллекцияОбъектов.Добавить();
    СтрокаКоллекции.Значение = Источник.Вес;
    СтрокаКоллекции.Свойство = Новый Структура("Наименование, ЭтоГруппа", "Вес", Ложь);
    СтрокаКоллекции.ТекстоваяСтрока = "";
    
    // объем номенклатуры
    СтрокаКоллекции = КоллекцияОбъектов.Добавить();
    СтрокаКоллекции.Значение = Источник.Объем;
    СтрокаКоллекции.Свойство = Новый Структура("Наименование, ЭтоГруппа", "Объем", Ложь);
    СтрокаКоллекции.ТекстоваяСтрока = "";
КонецЕсли;
Добавляем ПКО «СвойстваНоменклатуры» с пустым источником и приемником типа «ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения». В это ПКО добавляем два ПКС «Наименование» и «ЭтоГруппа» и по ним же устанавливаем флаги поиска.
Теперь работаем с ПКС, подчиненными ПКГС «ДополнительныеСвойства» ПКО «Номенклатура»: для ПКС «Значение» ничего не определяем, т.к. выгружаемые значения Веса и Объема являются значениями примитивного типа; для ПКС «Свойство» явно указываем, созданное ранее ПКО «СвойстваНоменклатуры».
3. Настраиваем механизм дополнительных свойств для конфигурации БП3 и добавляем два необходимых нам свойства. Однако для свойства «Объем» не достаточно разрядности после запятой, поэтому в конфигураторе БП3 для объекта ПВХ «ДополнительныеРеквизитыИСведения» увеличиваем точность числового типа значения до шести знаков.
4. Загружаем новые правила в макет плана обмена в конфигурации УТ11. Настраиваем on-line обмен для наших конфигураций и производим тестовые обмены – значения наших свойств корректно заполняются в БП3.

Остальное отправил на почту мастер группы.

Антон МухоморОтветить09.12.2012 в 17:43
1. При выполнении задания использовал УТ11.0.9.14(РФ) и БП3.0.16.6(РФ)
2. В настройках программы установил признак «Использовать дополнительные реквизиты и сведения»
3. В «Администрирование»программы, в «Обмен данными» установил флаг «Обмен данными»
4. В демо базе УТ11, с 12 апреля не выполнялся обмен. :) Выполнил обмен.
5. В Конфигураторе УТ, в ПВХ «Дополнительные реквизиты и сведения» добавил два реквизита Вес и Объем.
6. В Предприятии УТ, в Администрирование -> Общие настройки -> Наборы дополнительных реквизитов и сведений, включил использование Веса и Объема, как доп.реквизитов(Для Номенклатуры)
6. Сохранил в файл правила обмена между УТ и БП
7. Выгрузил структуру метаданных для УТ и БП и обновил ими уже имеющиеся конфигурации в КД
8. В КД добавил ПКО «ДополнительныеРеквизитыИСведения»
-Пустой Источник – Приемник(ПВХ:ДополнительныеРеквизитыИСведения)
-Установил флаг поиска по наименованию
-Установил флаг у «Не создавать новый объект в приемнике, если не найден»
9. Создал ПКС: Наименование и в Обработчике «ПередВыгрузкой» написал:

   Значение=Источник;
10. Создал ГПКС «Дополнительные реквизиты»
Пустой Источник – Приемник «Дополнительные реквизиты»
В Обработчике «ПередОбработкой» написал:

 Если Источник.ЭтоГруппа Тогда
      Отказ=Истина;
 Иначе
     ТабЗнач = Новый ТаблицаЗначений();
     ТабЗнач.Колонки.Добавить("Значение");
     ТабЗнач.Колонки.Добавить("Свойство");
     
     Вес=Источник.Вес;
     Если ЗначениеЗаполнено(Вес) Тогда
          Строка=ТабЗнач.Добавить();
          Строка.Значение=Вес;
          Строка.Свойство="Вес";
     Конецесли;

     Объем=Источник.Объем;
     Если ЗначениеЗаполнено(Объем) Тогда
          Строка = ТабЗнач.Добавить();
          Строка.Значение=Объем;
          Строка.Свойство="Объем";
     Конецесли;
 Конецесли;
11. Сохранил Правила обмена в файл.
12. В УТ загрузил правила в «тестовом» режиме.
13. Проверил выгрузку/загрузку.
14. В Конфигураторе УТ, в ветке Общие – > Планы обмена – >
ОбменУправлениеТорговлейБухгалтерияПредприятия30– > Макеты – > ПравилаОбмена, вставил сохраненные в КД правила обмена.

basОтветить09.12.2012 в 17:25
Практическое задание выполнил.
1. Подкорректировал типовые правила обмена УТ->БП. Добавил ПКГС для табличной части «Дополнительные реквизиты» справочника «Номенклатура». Для ПКГС в событии «Перед обработкой» вручную формирую таблицу значений с 2-мя строками (Вес и объем), сформированную таблицу значений присваиваю параметру «КоллекцияОбъектов». Таблица значений имеет 2 колонки: Свойство (строка) и Значение (Число). В колонку «Свойство» записывается строка «Вес» или «Объем» – соответственно, создал служебное ПКО для конвертации из строки в ПВХ. В ПКО используется содержится единственное ПКС «Наименование» с установленной галочкой «Поиск».
2. Предварительно в Бухгалтерии создал 2 элемента ПВХ с наименованиями «Вес», «Объем». Указал тип значения = Число, включил использование для справочника «Номенклатура».
3. В УТ создал новый обмен с БП, подгрузил модифицированные правила. Чтобы при изменении объема/веса элемент справочника Номенклатура регистрировался в плане обмена, снял галочку «Использовать фильтр выборочной регистрации объектов» в диалоге загрузки правил конвертации.

Составил ментальную карту (состоит из 2-х листов), отправил наmg@spec8.ru. В первом листе выписал для себя основные моменты в целом по курсу. Во втором листе составил примерный алгоритм работы on-line обмена исходя из своих наблюдений. Мне кажется, что-то подобное неплохо бы включить в содержание курса, т.к. после просмотра видеоуроков у меня не сложилось четкого понимания алгоритма работы on-line обмена (последовательность действий системы при загрузке/выгрузке; как осуществляется поиск ссылочных объектов и зачем нужен регистр соответствий; для меня было огромным удивлением обнаружить, что объекты не выгружаются по ссылкам и т.д.).

Отзыв по курсу оставлял ранее в соответствующей ветке. Ко всему сказанному добавлю: весьма удачно подобрано финальное практическое ДЗ, оно подтолкнуло меня к детальному разбору работы on-line обмена, да и к изучению функционала БСП вообще. Курс интересный, понятный, полезный. Выражаю благодарность авторам курса и желаю дальнейших успехов и развития!

АнастасияОтветить09.12.2012 в 15:48
Решение финального задания:
1. В БП 3.0 включила поддержку дополнительных реквизитов и свойств
2. Из УТ обмен данными выгрузила типовые правила обмена.
3. В КД на всякий случай обновила структуры метаданных УТ и БП и загрузила сохраненные правила обмена
4. Создала 2 ПКГС с пустым источником и приемником ТЧ «Дополнительные реквизиты» для веса и объема.
5. В обработчике «ПередОбработкой» прописала создание таблицы значений с соответствующими колонками «Свойство» и «Значение» и ее заполнение. КоллекцииОбъектов присвоила созданную таблицу значений.
6. Создала ПКС «Свойство»: источник – пустой, приемник – «Свойство» и ПКС «Значение»: источник – пустой, приемник – «Значение».
7. Создала новое ПКО для ПВХ «ДополнительныеРеквизитыИСведения» синхронизация по наименованию и генерация кода автоматически. В ПКС «Наименование» в обработчике «ПередВыгрузкой» прописала Значение = Источник;
8.Для определения типа значения ПВХ использовала прием с созданием узла «Типы». В ПКС «Свойство» ПКГС справочника «Номенклатура» определила ИсходящиеДанные и записала в них тип данных «Число». А в ПКО ПВХ создала новый узел Типы.
9. Сохранила правила в файл.
10. В Обмене данными УТ загрузила данные правила из файла. Выполнила выгрузку и загрузку.

Сергей КалмыковОтветить09.12.2012 в 14:14
1. Практическая задача
Зайдем в настройку обмена УТ11 и извлечем оттуда правила выгрузки данных.
Загрузим правила в конфигурацию «Конвертация данных».
В БП создадим дополнительные реквизиты «Вес» и «Объем» с числовым типом данных
Создадим ПКО Строка-ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения
Синхронизация по наименованию
Загрузим метаданные из БП 30, чтобы появились таб часть доп. реквизиты справочника «Номенклатура»
В ПКГС напишем код:

    Запрос = Новый Запрос;
    Запрос.Текст = 
    "ВЫБРАТЬ
    |   ""Вес"" КАК Свойство,
    |   Номенклатура.Вес КАК Значение,
    |   Номенклатура.Ссылка
    |ИЗ
    |   Справочник.Номенклатура КАК Номенклатура
    |
    |ОБЪЕДИНИТЬ ВСЕ
    |
    |ВЫБРАТЬ
    |   ""Объем"",
    |   Номенклатура.Объем,
    |   Номенклатура.Ссылка
    |ИЗ
    |   Справочник.Номенклатура КАК Номенклатура";
    
    
    Результат = Запрос.Выполнить();
    КоллекцияОбъектов=Новый ТаблицаЗначений;
    КоллекцияОбъектов.Колонки.Добавить("Свойство");
    КоллекцияОбъектов.Колонки.Добавить("Значение");
    
    Коллекция = Результат.Выгрузить();
    Для каждого Стр Из Коллекция Цикл
        Если Стр.Cсылка=Источник Тогда
            НоваяСтрока = КоллекцияОбъектов.Добавить();
            НоваяСтрока.Свойство = Стр.Свойство;
            НоваяСтрока.Значение = Стр.Значение;
        КонецЕсли;
    КонецЦикла;
Сохраним правила и загрузим их в УТ 11.
Проверим работу.
Далее загрузим правила в макет.
2. Ментальная карта
http://rusfolder.com/34011571
3. Отзыв
До курса по конвертации у меня был небольшой опыт переноса данных как
посредством текстовых файлов, ДБФ-файлов, кд версий 7.7 и 8
Я читал книжку «Конвертация данных».
Однако не было целостной картины функционала этой конфигурации.
Больше всего понравился 1-й раздел -масса практических примеров и практических домашних заданий. Есть мотивация выполнять задания – получаешь эталонное решение только после своего пусть даже не очень эффективного решения.
Еще до прохождения курса мне очень помогло первое бонусное видео.
Была по работе задача переносить данные очень старого и доработанного релиза УТ11 в актуальный релиз БП20. Разрабатывал правила с нуля.
Пользовался как примером правила из типовых. Возникали вопросы.
Теперь после прохождения курса не страшно браться ни за какие задачи по переносу.
Описание типовых универсальных механизмов помогает понять, как происходит перенос в типовых, где перенос интегрирован с БСП.
Все не запомнишь, но диск из физ. поставки и книжка – хороший справочник, который всегда под рукой.
Сейчас при переходе на БП30 ранее написанные переносы прийдется переделыветь, так что курс вышел как некогда вовремя.

dumalОтветить09.12.2012 в 13:29
Задание выполнено.
1. Установил ФО «Использовать дополнительные реквизиты и сведения»
2. Добавил два новых вида свойств – Вес и Объем, оба – число
3. Выгрузил правила обмена, добавил их в КД
4. Сначала пошел по неверному пути: Создал новое упрощенное ПКО по переносу номенклатуры, где указал ПКГС ДополнительныеСвойства и ПВД с таким кодом:

    ВыборкаДанных=Новый ТаблицаЗначений;
    ВыборкаДанных.Колонки.Добавить("Код");
    ВыборкаДанных.Колонки.Добавить("Наименование");
    ВыборкаДанных.Колонки.Добавить("ЭтоГруппа");
    ВыборкаДанных.Колонки.Добавить("Родитель");
    ВыборкаДанных.Колонки.Добавить("ДополнительныеРеквизиты");
    Запрос = Новый Запрос;
    Запрос.Текст = 
        "ВЫБРАТЬ
        |   Номенклатура.Код,
        |   Номенклатура.Наименование,
        |   Номенклатура.ЭтоГруппа,
        |   Номенклатура.Родитель,
        |   Номенклатура.Объем,
        |   Номенклатура.Вес
        |ИЗ
        |   Справочник.Номенклатура КАК Номенклатура
        |ГДЕ
        |   НЕ Номенклатура.ЭтоГруппа
        |   И (Номенклатура.Вес <> 0
        |           ИЛИ Номенклатура.Объем <> 0)";

    Результат = Запрос.Выполнить();

    ВыборкаДетальныеЗаписи = Результат.Выбрать();

    Пока ВыборкаДетальныеЗаписи.Следующий() Цикл
        СтрокаТЗ=ВыборкаДанных.Добавить();
        ЗаполнитьЗначенияСвойств(СтрокаТЗ,ВыборкаДетальныеЗаписи);
        
        ТЗДопСвойства=Новый ТаблицаЗначений;
        ТЗДопСвойства.Колонки.Добавить("Свойство");
        ТЗДопСвойства.Колонки.Добавить("Значение");
        ТЗДопСвойства.Колонки.Добавить("ТекстоваяСтрока");
        
        СтрокаДопСвойства=ТЗДопСвойства.Добавить();
        СтрокаДопСвойства.Свойство="Вес";
        СтрокаДопСвойства.Значение=ВыборкаДетальныеЗаписи.Вес;
        СтрокаДопСвойства.ТекстоваяСтрока=""+ВыборкаДетальныеЗаписи.Вес+" кг";
        
        СтрокаДопСвойства=ТЗДопСвойства.Добавить();
        СтрокаДопСвойства.Свойство="Объем";
        СтрокаДопСвойства.Значение=ВыборкаДетальныеЗаписи.Объем;
        СтрокаДопСвойства.ТекстоваяСтрока=""+ВыборкаДетальныеЗаписи.Объем+" м3";

        СтрокаТЗ.ДополнительныеРеквизиты=ТЗДопСвойства;
    КонецЦикла;
В свойствах ПКС «Свойство» ТЧ в обработчике ПриВыгрузке указал следующую строку:
ТипПриемника=»ПланВидовХарактеристикСсылка.ДополнительныеРеквизитыИСведения»;

Но этот подход не сработал, поэтому пошел по другому пути

5. В оригинальном ПКО по переносу номенклатуры добавил два новых ПКС с переносом значений в параметр, после чего в обработчике ПослеЗагрузки ПКО Номенклатура добавил такой код:

Если ПараметрыОбъекта<>Неопределено Тогда
    Если ПараметрыОбъекта.Получить("Вес")<>Неопределено Тогда
        Если ПараметрыОбъекта.Получить("Вес")<>0 Тогда
            Вес=ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Вес");
            Строка=Объект.ДополнительныеРеквизиты.Найти(Вес.Ссылка,"Свойство");
            Если Строка=Неопределено Тогда
                Строка=ОБъект.ДополнительныеРеквизиты.Добавить();
                Строка.Свойство=Вес;
            КонецЕсли;
            Строка.Значение=ПараметрыОбъекта.Получить("Вес");
            Строка.ТекстоваяСтрока=""+Строка.Значение+" кг.";
        КонецЕсли;
    КонецЕсли;
КонецЕсли;

Если ПараметрыОбъекта<>Неопределено Тогда
    Если ПараметрыОбъекта.Получить("Объем")<>Неопределено Тогда
        Если ПараметрыОбъекта.Получить("Объем")<>0 Тогда
            Объем=ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Объем");
            Строка=Объект.ДополнительныеРеквизиты.Найти(Объем.Ссылка,"Свойство");
            Если Строка=Неопределено Тогда
                Строка=ОБъект.ДополнительныеРеквизиты.Добавить();
                Строка.Свойство=Объем;
            КонецЕсли;
            Строка.Значение=ПараметрыОбъекта.Получить("Объем");
            Строка.ТекстоваяСтрока=""+Строка.Значение+" м3";
        КонецЕсли;
    КонецЕсли;
КонецЕсли;
Ручной перенос заработал. После этого делом техники было добавить обновленные правила в типовой план обмена. Полагаю, мое решение является грубо неоптимальным, однако первый подход, к сожалению, не сработал. Буду рад увидеть типовое решение Евгения

АлександрОтветить09.12.2012 в 13:20
Здравствуйте.
Отчет по финальному заданию.
Решение практической задачи.
Опять таки напомню что я Из Казахстана, поэтому производил перенос из Бухгалтерии 2.0 в УТ 3.0. Переносил я реквизиты: Код ТНВЫД и Код КПВЭД, я не думаю что от этого сильно поменялся принцип задания. В УТ 3.0 в настройках установил галочку использования дополнительных реквизитов. Модифицировал правила обмена, добавил ПКГС для номенклатуры определил обработчик перед обработкой:

Если Не Источник.ЭтоГруппа Тогда
    КоллекцияОбъектов = Новый ТаблицаЗначений;
    КоллекцияОбъектов.Колонки.Добавить("Значение");
    КоллекцияОбъектов.Колонки.Добавить("Свойство");
    КоллекцияОбъектов.Колонки.Добавить("ТекстоваяСтрока");
    Если Источник.КодКПВЭД <> "" Тогда
        СтрокаДопСвойств = КоллекцияОбъектов.Добавить();
        СтрокаДопСвойств.Значение = Источник.КодКПВЭД;
        СтрокаДопСвойств.Свойство = "КодКПВЭД";
        СтрокаДопСвойств.ТекстоваяСтрока = "Код КПВЭД";
    КонецЕсли;
    Если Источник.КодТНВЭД <> "" Тогда
        СтрокаДопСвойств = КоллекцияОбъектов.Добавить();
        СтрокаДопСвойств.Значение = Источник.КодТНВЭД;
        СтрокаДопСвойств.Свойство = "КодТНВЭД";
        СтрокаДопСвойств.ТекстоваяСтрока = "Код ТНВЭД";
    КонецЕсли;
КонецЕсли;
Создал ПКО для плана видов характеристик «Дополнительные реквизиты и сведения». Проблемы как всегда возникли при переносе типа значения. Что я только не делал в обработчиках, и добавлял узел типы и после загрузки присваивал определенный тип, ничего не происходила. А проблема оказалась в том что при не явной выгрузки объекта программа нивкакую не выполняла обработчики. Пришлось явно выгрузить эти 2 элемента воспользовавшись экспортной процедурой в обработчике перед выгрузкой данных КД:

ВыгрузитьПоПравилу("КодКПВЭД",,,,"ДополнительныеРеквизитыИСведения",,,,,,,,Ложь);
ВыгрузитьПоПравилу("КодТНВЭД",,,,"ДополнительныеРеквизитыИСведения",,,,,,,,Ложь);
Тогда то мой обработчик после загрузки элемента дополнительных реквизитов сработал корректно:

КС = Новый КвалификаторыСтроки(20);
Массив = Новый Массив;
Массив.Добавить(Тип("Строка"));
ОписаниеТиповС = Новый ОписаниеТипов(Массив, , КС);


Объект.ТипЗначения = ОписаниеТиповС;
P.S. Также после выгрузки добавил этот элемент в набор дополнительных реквизитов номенклатуры:

НаборНоменклатуры = Справочники.НаборыДополнительныхРеквизитовИСведений.Справочник_Номенклатура_Общие;
Отбор = Новый Структура;
Отбор.Вставить("Свойство", Объект.Ссылка);
Если Не НаборНоменклатуры.ДополнительныеРеквизиты.НайтиСтроки(Отбор).Количество() Тогда
    НаборНоменклатуры = НаборНоменклатуры.ПолучитьОбъект();
    НовыяСтрокаНабора = НаборНоменклатуры.ДополнительныеРеквизиты.Добавить();
    НовыяСтрокаНабора.Свойство = Объект.Ссылка;
    НаборНоменклатуры.Записать();
КонецЕсли;
Далее по финальному заданию:
Отзыв по курсу я отправлял ранее, правда фотографию не отправлял, в ближайшее время отправлю.

Ментальную карту туда же отправлять куда и отзыв?

Андрей ШнитовОтветить10.12.2012 в 05:10
Здравствуйте, Александр!
Ментальную карту и фотографию отправляйте на mg@spec8.ru

mbreakerОтветить09.12.2012 в 12:44
Выгрузила типовые правила обмена из конфигурации УТ, загрузила их в Конвертацию данных.

Выгрузила описание метаданных УТ и БП, загрузила в конвертацию.

В ПКО Номенклатура добавила ПКГС ДополнительныеРеквизиты с флагом «Получить из входящих данных».

В обработчике «Перед выгрузкой» ПКО Номенклатура:

Если НЕ Источник.ЭтоГруппа Тогда

    ТаблицаДопРеквизиты = Новый ТаблицаЗначений;
    ТаблицаДопРеквизиты.Колонки.Добавить("Свойство");
    ТаблицаДопРеквизиты.Колонки.Добавить("Значение");

    СтрокаВес = ТаблицаДопРеквизиты.Добавить();
    СтрокаВес.Свойство = "Вес";
    СтрокаВес.Значение = Источник.Вес;
    СтрокаОбъем = ТаблицаДопРеквизиты.Добавить();
    СтрокаОбъем.Свойство = "Объем";
    СтрокаОбъем.Значение = Источник.Объем;

    ВходящиеДанные.Вставить("ДополнительныеРеквизиты", ТаблицаДопРеквизиты);

КонецЕсли;
В обработчике «Перед обработкой» ПКГС ДополнительныеРеквизиты:

Если Источник.ЭтоГруппа Тогда
    Отказ = Истина;
КонецЕсли;
Создала ПКО ДополнительныеРеквизитыСтрока, источник – пустой, приемник – ПланВидовХарактеристик.ДополнительныеРеквизитыИСведения. Добавила ПКС Наименование (в обработчике «Перед выгрузкой»: Значение = Источник) и ПКС ЭтоГруппа, установила для них флаг поиска.
Установила это ПКО в качестве правила конвертации для ПКС Свойство ПКГС ДополнительныеРеквизиты.

Создала новый обмен в УТ, загрузила новые правила конвертации.

В БП в настройках программы установила флаг «Использовать дополнительные реквизиты», создала дополнительные реквизиты «Вес» и «Объем».

Продолжила создание обменав в БП. Выполнила выгрузку/загрузку данных.

